Matricinio lošimo sprendimas

Su komandomis vlow ir vupp apskaičiuojame lošimo apatinius ir viršutinius rėžius.

```
(%i1) vlow(A):=block([rows,cols],
      [rows,cols]:matrix_size(A),
      lmax(makelist(lmin(A[i]),i,1,rows)))$
```

```
(%i2) vupp(A):=block([rows,cols],
      [rows,cols]:matrix_size(A),
      lmin(makelist(lmax(transpose(A)[i]),i,1,cols)))$
```

A.Apynis, Lošimų teorija, VU, 2007, 6.1 pavyzdžio sprendimas.

Išspręskite matricinį lošimą, kai lošimo matrica yra

```
(%i3) A:matrix([6,3,1,-1],[-2,1,0,5]);
```

(%o3) $\begin{bmatrix} 6 & 3 & 1 & -1 \\ -2 & 1 & 0 & 5 \end{bmatrix}$

```
(%i4) vlow(A);
```

(%o4) -1

```
(%i5) vupp(A);
```

(%o5) 1

```
(%i6) Ap:A+3;
```

(%o6) $\begin{bmatrix} 9 & 6 & 4 & 2 \\ 1 & 4 & 3 & 8 \end{bmatrix}$

```
(%i7) load(simplex);
```

(%o7)
C:/Program Files (x86)/Maxima-sbcl-5.36.1/share/maxima/5.36.1/share/simplex/simplex.mac

```
(%i8) f1:u1+u2;
```

(%o8) $u2+u1$

```
(%i9) apr1:[9*u1+u2>=1,6*u1+4*u2>=1,4*u1+3*u2>=1,2*u1+8*u2>=1,u1>=0,u2>=0];
```

(%o9) $[u2+9\,u1>=1, 4\,u2+6\,u1>=1, 3\,u2+4\,u1>=1, 8\,u2+2\,u1>=1, u1>=0, u2>=0]$

```
(%i10) sol:minimize_lp(f1,apr1),nonegalive_lp=true;
```

(%o10) $\left[\frac{7}{26}, \left[u2=\frac{1}{13}, u1=\frac{5}{26}\right]\right]$

```
(%i11) s:subst(sol[2],[u1,u2]);
```

(%o11) $\left[\frac{5}{26}, \frac{1}{13}\right]$

```
(%i12) d:s[1]+s[2];
```

(%o12) $\frac{7}{26}$

Pirmojo lošėjo strategija yra:

```
(%i13) s1:s/d;
```

(%o13) $\left[\frac{5}{7}, \frac{2}{7}\right]$

```
(%i14) f2:w1+w2+w3+w4;
```
(%o14) $w4+w3+w2+w1$

```
(%i15) apr2:[9*w1+6*w2+4*w3+2*w4<=1,w1+4*w2+3*w3+8*w4<=1,w1>=0,w2>=0,w3>=0,w4>=0];
```
(%o15) $[2\ w4+4\ w3+6\ w2+9\ w1<=1, 8\ w4+3\ w3+4\ w2+w1<=1, w1>=0, w2>=0, w3>=0, w4>=0]$

```
(%i16) sol2:maximize_lp(f2,apr2);
```
(%o16) $[\frac{7}{26}, [w4=\frac{1}{26}, w3=\frac{3}{13}, w2=0, w1=0]]$

```
(%i17) s:subst(sol2[2],[w1,w2,w3,w4]);
```
(%o17) $[0, 0, \frac{3}{13}, \frac{1}{26}]$

```
(%i18) d:s[1]+s[2]+s[3]+s[4];
```
(%o18) $\frac{7}{26}$

Antrojo lošėjo strategija yra:

```
(%i19) s2:s/d;
```
(%o19) $[0, 0, \frac{6}{7}, \frac{1}{7}]$

Lošimo vertė yra:

```
(%i20) s1.A.s2;
```
(%o20) $\frac{5}{7}$