Lokalieji ekstremumai

A.Domarkas, VU

Komanda "local_extr" randa n kintamųjų(n>=2) funkcijos f lokaliuosius ekstremumus.
Randami funkcijos hesianas ir pagrindiniai minorai.

```
(%i1) local_extr(f):=block([vars,n,kt,M,M1,realonly:true],local(A),
      vars:listofvars(f),
      n:length(vars),
      solve(makelist(diff(f,vars[k]),k,1,n),vars),
      kt:map(sort,%%),
      A(k):=hessian(f,makelist(vars[i],i,1,k)),
      M:makelist(determinant(A(k)),k,1,n),
      M1:makelist((-1)^i*M[i],i,1,n),
      for k thru length(kt) do
      if lmin(ev(M,kt[k]))>0 then print(kt[k],"- local minimum, ",f_min=subst(kt[k],f))
      elseif  lmin(ev(M1,kt[k]))>0 then print(kt[k],"- local maximum, ",f_max=subst(kt[k],f))
      elseif (n=2 and ev(M[n],kt[k])<0) or freeof(0,ev(M,kt[k])) then  print(kt[k],"- saddle poi
      else print(kt[k],"- we must examine the critical point by some other means")
      )$
```

1.

```
(%i2) f:x^2+y^2+(z+1)^2-x*y+x;
```

(%o2) $(z+1)^2+y^2-x\,y+x^2+x$

```
(%i3) local_extr(f);
```

$[x=-\frac{2}{3}, y=-\frac{1}{3}, z=-1]$ - *local minimum,* $f\_min=-\frac{1}{3}$

(%o3) *done*

2.

```
(%i4) f:x^4-y^4;
```

(%o4) x^4-y^4

```
(%i5) local_extr(x^4-y^4);
```

$[x=0, y=0]$ - *we must examine the critical point by some other means*

(%o5) *done*

Hesianas kritiniame taške [0,0] yra nulinė matrica. Todėl čia atsakymas negaunamas.
Tolimesniam tyrimui reikia nagrinėti aukštesnės eilės(>=3) išvestines.

```
(%i6) H:hessian(f,[x,y]);
```

(%o6) $\begin{bmatrix} 12\,x^2 & 0 \\ 0 & -12\,y^2 \end{bmatrix}$

```
(%i7) subst([x=0,y=0],H);
```

(%o7) $\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$

3.

```
(%i8) f:x^4+y^4;
```

(%o8) y^4+x^4

```
(%i9) local_extr(x^4+y^4);
```

$[x=0, y=0]$ - *we must examine the critical point by some other means*

(%o9) *done*

Hesianas kritiniame taške [0,0] yra nulinė matrica. Todėl čia atsakymas negaunamas.
Tolimesniam tyrimui reikia nagrinėti aukštesnės eilės(>=3) išvestines.

(%i10) `H:hessian(f,[x,y]);`

(%o10) $\begin{bmatrix} 12\,x^2 & 0 \\ 0 & 12\,y^2 \end{bmatrix}$

(%i11) `subst([x=0,y=0],H);`

(%o11) $\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$

4.

(%i12) `local_extr(x*y);`

$[x=0,y=0]$ - *saddle point*

(%o12) *done*

5.

(%i13) `local_extr(x^3/3-x-y^3/3+y);`

$[x=1,y=1]$ - *saddle point*

$[x=-1,y=1]$ - *local maximum,* $f\_max=\frac{4}{3}$

$[x=1,y=-1]$ - *local minimum,* $f\_min=-\frac{4}{3}$

$[x=-1,y=-1]$ - *saddle point*

(%o13) *done*

6.

(%i14) `local_extr(x^3+3*x*y^2-39*x-36*y+26)$`

$[x=3,y=2]$ - *local minimum,* $f\_min=-100$

$[x=2,y=3]$ - *saddle point*

$[x=-2,y=-3]$ - *saddle point*

$[x=-3,y=-2]$ - *local maximum,* $f\_max=152$

7. Stewart, Calculus, 7ed, sec. 14.7, example 4

(%i15) `fpprintprec:6$`

(%i16) `f:10*x^2*y-5*x^2-4*y^2-x^4-2*y^4;`

(%o16) $-2\,y^4-4\,y^2+10\,x^2\,y-x^4-5\,x^2$

(%i17) `local_extr(f)$`

$[x=0,y=0]$ - *local maximum,* $f\_max=0$

$[x=-2.64422,y=1.89838]$ - *local maximum,* $f\_max=8.49586$

$[x=-0.856657,y=0.646772]$ - *saddle point*

$[x=0.856657,y=0.646772]$ - *saddle point*

$[x=2.64422,y=1.89838]$ - *local maximum,* $f\_max=8.49586$

Detalus sprendimas:

(%i18) `load(odes)$`

(%i19) `eq1:diff(f,x)=0;`
`eq2:diff(f,y)=0;`

(%o19) $20\,x\,y-4\,x^3-10\,x=0$

(%o20) $-8\,y^3-8\,y+10\,x^2=0$

Aišku, kad [x=0, y=0] yra šios sistemos sprendinys.
Todėl eq1 dalijame iš x ir išreiškiame x^2 :

```
(%i21) solve(eq1/x,x^2);
```
(%o21) $[x^2=\frac{10\,y-5}{2}]$

```
(%i22) eq3:expand(%[1]);
```
(%o22) $x^2=5\,y-\frac{5}{2}$

```
(%i23) solx:[x=sqrt(rhs(%)),x=-sqrt(rhs(%))];
```
(%o23) $[x=\sqrt{5\,y-\frac{5}{2}},x=-\sqrt{5\,y-\frac{5}{2}}]$

```
(%i24) eq4:subst(eq3,eq2),expand;
```
(%o24) $-8\,y^3+42\,y-25=0$

```
(%i25) soly:solvet(eq4,y);
```
(%o25) $[y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-3\,\pi}{3}\right),y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-\pi}{3}\right),y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)+\pi}{3}\right)]$

```
(%i26) float(%), numer;
```
(%o26) $[y=-2.54516,y=1.89838,y=0.646772]$

Matome, kad šios reikšmės sutampa su anksčiau apskaičiuotomis.
Pirmasis sprendinys iš soly netinka, nes įstačius į solx gaunami menami sprendinai:

```
(%i27) subst(soly[1],solx)$
       float(rectform(%));
```
(%o28) $[x=3.90202\,\%i,x=-3.90202\,\%i]$

Gauname, kad kritinių taškų tikslios išraiškos yra:

```
(%i29) kt:disp([0,0],[ev(solx[1],soly[2]),soly[2]],[ev(solx[2],soly[2]),soly[2]],
              [ev(solx[1],soly[3]),soly[3]],[ev(solx[2],soly[3]),soly[3]])$
```
$[0,0]$

$$[x=\sqrt{5\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-\pi}{3}\right)-\frac{5}{2}},y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-\pi}{3}\right)]$$

$$[x=-\sqrt{5\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-\pi}{3}\right)-\frac{5}{2}},y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)-\pi}{3}\right)]$$

$$[x=\sqrt{5\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)+\pi}{3}\right)-\frac{5}{2}},y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)+\pi}{3}\right)]$$

$$[x=-\sqrt{5\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)+\pi}{3}\right)-\frac{5}{2}},y=\sqrt{7}\cos\left(\frac{\operatorname{atan}\left(\frac{3\sqrt{83}}{25}\right)+\pi}{3}\right)]$$

Kritinių taškų išraiškas galima supaprastinti įvedus pažymėjimą:

```
(%i30) tr:omega=atan((3*sqrt(83))/25)/3$
```

Pavyzdžiui:

```
(%i31) subst(reverse(tr),expand(soly[2]));
```
(%o31) $y=\sqrt{7}\cos\left(\omega-\frac{\pi}{3}\right)$

Pavyzdžio pabaigai nubrėšime funkcijos f lygio linijas:

(%i32) load(implicit_plot)$

(%i33) wximplicit_plot([f=3,f=7,f=-1,f=-3/2,f=-10,f=-20,f=-30],
[x,-4,4],[y,-2,3],[legend,false]);

(%t33)



(%o33)

8 pvz

(%i34) f:(x-1/2)^2*(x+1)^2+(y+1)^2*(y-1)^2;

(%o34) $(y-1)^2\,(y+1)^2+\left(x-\frac{1}{2}\right)^2\,(x+1)^2$

(%i35) local_extr(f);

$[x=\frac{1}{2}, y=0]$ - *saddle point*

$[x=-1, y=0]$ - *saddle point*

$[x=-\frac{1}{4}, y=0]$ - *local maximum*, $f\_max=\frac{337}{256}$

$[x=\frac{1}{2}, y=1]$ - *local minimum*, $f\_min=0$

$[x=-1, y=1]$ - *local minimum*, $f\_min=0$

$[x=-\frac{1}{4}, y=1]$ - *saddle point*

$[x=\frac{1}{2}, y=-1]$ - *local minimum*, $f\_min=0$

$[x=-1, y=-1]$ - *local minimum*, $f\_min=0$

$[x=-\frac{1}{4}, y=-1]$ - *saddle point*

(%o35) *done*

(%i36) 337/256;

(%o36) $\frac{337}{256}$

(%i37) float(%), numer;

(%o37) 1.31641

(%i38) load(implicit_plot)$

(%i39) ratprint:false$

(%i40)
```
wximplicit_plot([f=0.03,f=0.1,f=81/256,f=0.5,f=1,f=1.1,f=1.3,f=1.5],
[x,-1.6,1.2],[y,-1.5,1.5],[legend,false],[box,false]),wxplot_size=[600,400];
```

(%t40)



(%o40)

(%i57)
```
load(draw)$
```

(%i42)
```
set_draw_defaults(
          xrange = [-2,1.5],
          yrange = [-1.5,2],
          grid   = true,
          proportional_axes = xy,
          fill_color = skyblue)$
```

(%i43)
```
with_slider_draw(
         z, makelist(0.05*i, i, 1, 33),
         key    = string(f=z),
         implicit(f=z,x,-2,1.5,y, -1.5, 2));
```

(%t43)



(%o43)

9 pvz, Demidovič 3642

(%i44)
```
f:x^2+y^2+z^2+2*x+4*y-6*z;
```

(%o44) $z^2-6\,z+y^2+4\,y+x^2+2\,x$

```
(%i45) local_extr(f);
```
$[x=-1,y=-2,z=3]$ - *local minimum,* $f\_min=-14$

(%o45) *done*

10 pvz, Demidovič 3643

```
(%i46) f:x^3+y^2+z^2+12*x*y+2*z;
```
(%o46) $z^2+2\,z+y^2+12\,x\,y+x^3$

```
(%i47) local_extr(f);
```
$[x=24,y=-144,z=-1]$ - *local minimum,* $f\_min=-6913$

$[x=0,y=0,z=-1]$ - *we must examine the critical point by some other means*

(%o47) *done*

```
(%i48) eq1:diff(f,x)=0;
       eq2:diff(f,y)=0;
       eq3:diff(f,z)=0;
```
(%o48) $12\,y+3\,x^2=0$

(%o49) $2\,y+12\,x=0$

(%o50) $2\,z+2=0$

```
(%i51) kt:solve([eq1,eq2,eq3]);
```
(%o51) $[[z=-1,y=-144,x=24],[z=-1,y=0,x=0]]$

Pagrindiniai minorai

```
(%i52) M1:determinant(hessian(f,[x]));
```
(%o52) $6\,x$

```
(%i53) M2:determinant(hessian(f,[x,y]));
```
(%o53) $12\,x-144$

```
(%i54) M3:determinant(hessian(f,[x,y,z]));
```
(%o54) $24\,x-288$

```
(%i55) subst(kt[1],[M1,M2,M3]);
```
(%o55) $[144,144,288]$

```
(%i56) subst(kt[2],[M1,M2,M3]);
```
(%o56) $[0,-144,-288]$