Dviejų kintamųjų funkcijų ekstremumai

A.Domarkas, VU

Žr. [1], 210 p.; [2], 218 p.; [3]

Pavyzdys(žr. [2]). Rasime funkcijos f ekstremumus.

```
(%i1) f:x^3+y^3-3*x^2-3*y^2-9*x;
```

(%o1) $y^3-3\,y^2+x^3-3\,x^2-9\,x$

```
(%i2) L1:diff(f,x)=0; L2:diff(f,y)=0;
```

(%o2) $3\,x^2-6\,x-9=0$

(%o3) $3\,y^2-6\,y=0$

Randame kritinius taškus:

```
(%i4) krit_taskai:solve([L1,L2],[x,y]);
```

(%o4) $[[x=3,y=0],[x=-1,y=0],[x=3,y=2],[x=-1,y=2]]$

Kritiniams taškams tirti apibrėžiame komandą "testas":

```
(%i5) testas(kt):=block(
      [a,b,c]:subst(kt,[diff(f,x,2),diff(f,x,1,y,1),diff(f,y,2)]),
      if a*c-b^2>0 and a<0 then print("Taške ",kt,maksimumas=subst(kt,f))
      elseif a*c-b^2>0 and a>0 then print("Taške ",kt,minimumas=subst(kt,f)
      elseif a*c-b^2<0 then print("Taške ",kt,"ekstremumo nėra(balno taškas
      elseif a*c-b^2=0 then print("Taške ",kt,"reikia papildomo tyrimo"))$
```

```
(%i6) map(testas,krit_taskai)$
```

Taške $[x=3,y=0]$ *ekstremumo nėra(balno taškas)*

Taške $[x=-1,y=0]$ $maksimumas=5$

Taške $[x=3,y=2]$ $minimumas=-31$

Taške $[x=-1,y=2]$ *ekstremumo nėra(balno taškas)*

```
(%i7) wxcontour_plot(f, [x, -2, 4], [y, -1.5, 3.5], [legend,false],
      [gnuplot_preamble,"set cntrparam levels 24"])$
```

(%t7)



2 pavyzdys. (žr. [3])

```
(%i8) f:(x+y)*(x*y+x*y^2);
```

(%o8) $(y+x)(x\,y^2+x\,y)$

```
(%i9) L1:diff(f,x)=0; L2:diff(f,y)=0;
```

(%o9) $(y+x)(y^2+y)+x\,y^2+x\,y=0$

(%o10) $x\,y^2+(y+x)(2\,x\,y+x)+x\,y=0$

Randame kritinius taškus:

```
(%i11) krit_taskai:solve([L1,L2],[x,y]);
```

(%o11) $[[x=0,y=0],[x=0,y=-1],[x=\frac{3}{8},y=-\frac{3}{4}],[x=1,y=-1]]$

(%i12) `map(testas,krit_taskai)$`

Taške $[x=0, y=0]$ *reikia papildomo tyrimo*

Taške $[x=0, y=-1]$ *ekstremumo nėra(balno taškas)*

Taške $[x=\frac{3}{8}, y=-\frac{3}{4}]$ *maksimumas*$=\frac{27}{1024}$

Taške $[x=1, y=-1]$ *ekstremumo nėra(balno taškas)*

(%i13) `wxcontour_plot(f, [x, 0, 1], [y, -1, 0], [legend,false], [gnuplot_preamble,"set cntrparam levels 36"])$`

(%t13)



3 pavyzdys. Rasime Himmelblau funkcijos([4]) ekstremumus.

(%i14) `f:(x^2+y-11)^2+(x+y^2-7)^2;`

(%o14) $(y^2+x-7)^2+(y+x^2-11)^2$

(%i15) `L1:diff(f,x)=0; L2:diff(f,y)=0;`

(%o15) $2(y^2+x-7)+4x(y+x^2-11)=0$

(%o16) $4y(y^2+x-7)+2(y+x^2-11)=0$

Randame kritinius taškus:

```
(%i17) krit_taskai:solve([L1,L2],[x,y]);
(%o17) [[x=3.584428223844282,y=-1.848126535626536],[x=-2.80511811023622,y=3.131312515247621],[x=-3.779310344827586,y=-3.283185840707965],[x=3,y=2],[x=0.086677504919997,y=2.884254431699687],[x=-0.12796134663342,y=-1.953714981729598],[x=-0.27084458851071,y=-0.92303848075962],[x=-3.073025757146901,y=-0.081353045901386],[x=3.38515406162465,y=0.073851882498966]]
```

```
(%i18) length(%);
(%o18) 9
```

Gavome, kad fukcija turi 9 kritinius taškus. Jų klasifikacijai pritaikome komandą "testas".

```
(%i19) map(testas,krit_taskai)$
```

Taške [x=3.584428223844282,y=-1.848126535626536] *minimumas*= $7.190238265563075910^{-13}$

Taške [x=-2.80511811023622,y=3.131312515247621] *minimumas*= $1.805914487428135810^{-14}$

Taške [x=-3.779310344827586,y=-3.283185840707965] *minimumas*= $1.875485921124983410^{-12}$

Taške [x=3,y=2] *minimumas*=0

Taške [x=0.086677504919997,y=2.884254431699687] *ekstremumo nėra(balno taškas)*

Taške [x=-0.12796134663342,y=-1.953714981729598] *ekstremumo nėra(balno taškas)*

Taške [x=-0.27084458851071,y=-0.92303848075962] *maksimumas*= 181.6165215225827

Taške [x=-3.073025757146901,y=-0.081353045901386] *ekstremumo nėra(balno taškas)*

Taške [x=3.38515406162465,y=0.073851882498966] *ekstremumo nėra(balno taškas)*

Brėžime lygio linijas vienintelio maksimumo taško [x=-0.27084458851071,y=-0.92303848075962] aplinkoje:

```
(%i20) wxcontour_plot(f, [x,-0.3,-0.24], [y,-1,-0.85], [legend,false],
       [gnuplot_preamble,"set cntrparam levels 12"])$
```

(%t20)



Geriau už "contour_plot" nubrėžti lygio linijas galima su paketu "implicit_plot".
Tada lygių reikšmes galima parinkti patiems.

```
(%i21) load(implicit_plot)$
```

```
(%i22) wximplicit_plot([f=1,f=10,f=20,f=30,f=160,f=180],
       [x,-5,5],[y,-5,5],[legend,false]);
```

(%t22)



(%o22)

Dar kartą randame visus penkis lokaliuosius ekstremumus su paketu "fmin_cobyla":

```
(%i29) load(fmin_cobyla)$
```

```
(%i24) fmin_cobyla(f, [x, y], [-3,3], constraints = [], iprint=0);
```

(%o24) $[[x=-2.805117979821587, y=3.131312918177591], 6.8615465162678513\ 10^{-12}, 54, 0]$

```
(%i25) fmin_cobyla(f, [x, y], [-3,-3], constraints = [], iprint=0);
```

(%o25) $[[x=-3.779310861835342, y=-3.283185474940021], 4.2159620383287268\ 10^{-11}, 53, 0]$

```
(%i26) fmin_cobyla(f, [x, y], [1,1], constraints = [], iprint=0);
```

(%o26) $[[x=3.000000072641843, y=1.99999958949148], 2.4636337057088336\ 10^{-12}, 58, 0]$

```
(%i27) fmin_cobyla(f, [x, y], [1,-2], constraints = [], iprint=0);
```

(%o27) $[[x=3.584427865893345, y=-1.848126721786767], 1.2991529721307269\ 10^{-11}, 58, 0]$

```
(%i28) fmin_cobyla(-f, [x, y], [-1,-1], constraints = [], iprint=0);
```

(%o28) $[[x=-0.27084518594943, y=-0.92303816154376], -181.6165215225746, 50, 0]$

Literatūra:

[1] V.Pekarskas, Trumpas matematikos kursas, Kaunas, Technologija, 2005,
[2] V.Būda, Matematiniai ekonominės analizės pagrindai, TEV, 2008
[3] http://en.wikipedia.org/wiki/Second_partial_derivative_test
[4] http://en.wikipedia.org/wiki/Himmelblau%27s_function