



Extreme points



A.Domarkas, VU, 2015



aleksas.domarkas@mif.vu.lt



In mathematics, an extreme point of a convex set S in a real vector space is a point in S which does not lie in any open line segment joining two points of S .

Intuitively, an extreme point is a "vertex" of S . See

http://en.wikipedia.org/wiki/Extreme_point

http://ljk.imag.fr/membres/Anatoli.Iouditski/cours/convex/chapitre_1.pdf



We use maxima 5.35.1.2

<http://sourceforge.net/projects/maxima/files/Maxima-Windows/5.35.1.2-Windows/>



```
(%i1) load(simplex)$
```



```
(%i2) ext(apr):=block([var,fs,cs,ap,s,S,m],
  var:sort(listofvars(apr)),
  s:apply("+",var),
  fs:append([1,s,-s],var,-var),
  ap(k):=subst(apr[k]=(lhs(apr[k])=rhs(apr[k])),apr),
  cs:makelist(ap(k),k,1,length(apr)),
  S:[],
  for f in fs do
  for c in cs do
  (
    m:minimize_lp(f,c),
    if listp(m) then
      S:cons(subst(m[2],var),S)
  ),
  listify(setify(S))
)$
```



1.



```
(%i3) apr:[x1+x2<=12, 2*x1-x2<=12,2*x1-x2>=0,2*x1+x2>=4,x2>=0];
```



```
(%o3) [ x2+x1<=12, 2 x1-x2<=12, 2 x1-x2>=0, x2+2 x1>=4, x2>=0 ]
```



```
(%i4) ext(apr);
```




```
(%o4) [[1,2],[2,0],[4,8],[6,0],[8,4]]
```



```
(%i14) load(draw)$
```

```
(%i6) wxdraw2d(
    fill_color=grey,
    grid=true,
    x_voxel = 50,
    y_voxel = 50,
    region(sr:apply("and",apr),x1,-1,10,x2,-1,10))$
```


```
(%t6)
```



2.

Figure 1:

$$0 \leq x_1 \leq 2, \quad 0 \leq x_2 \leq 2, \quad 0 \leq x_3 \leq 2, \quad x_1 + x_2 + x_3 \leq 5$$



```
(%i7) apr:[x[1]>=0,x[2]>=0,x[3]>=0,
    x[1]<=2,x[2]<=2,x[3]<=2,
    x[1]+x[2]+x[3]<=5]$
```

```
(%i8) ext(apr);
```

```
(%o8) [[0,0,0],[0,0,2],[0,2,0],[0,2,2],[1,2,2],[2,0,0],[2,0,2],[2,1,2],[2,2,0],[2,2,1]]
```

```
(%i9) length(%);  
(%o9) 10
```

3.

```
(%i10) apr:[y1-y2<=4,-y1+2*y2-3*y3<=2,3*y1+y2-y3<=0,  
y1+3*y2+3*y3<=-2,4*y1-y3<=5,y3<=3,3*y1+y2+3*y3<=6]$
```

```
(%i11) ext(apr);
```

```
(%o11) [[-11,0,3],[-1/5,0,-3/5],[2/11,-42/11,-36/11],[1/4,-15/4,3],[11/8,-21/8,3/2]]
```

4.

```
(%i12) apr:[-x[1]<=0,-x[2]<=0,-x[3]<=0,  
-x[2]-x[1]+4/5<=0,-x[3]-x[1]+2/5<=0,  
-x[3]-x[2]+1/5<=0,x[3]+x[2]+x[1]=1]$
```

```
(%i13) ext(apr);
```

```
(%o13) [[1/5,3/5,1/5],[2/5,3/5,0],[4/5,0,1/5],[4/5,1/5,0]]
```