Lagranžo daugiklių metodas

Čia yra naudojama atvirojo kodo kompiuterines algebros sistema maxima 5.31.2.
Jos atskiras versijas galite rasti Windows, Linux, Mac OS ir Android sistemose.

Sąlyginio ekstremumo uždaviniams yra naudojamas Lagranžo daugiklių metodas.
Teoriją žr. [1], 14.8, Lagrange Multipliers, p. 966; [4], p. 110; [5]
Sąlyginio absoliutaus ekstremumo uždaviniams spręsti apibrėžiame komandas:

```
(%i1) minimize_lagr(f,apr):=block([n,n1,L,l,v,v1,m,sist,s,sv,spr],
      v:listofvars([f,apr]),
      n:length(v),
      n1:length(apr),
      L:f+sum(l[k]*(lhs(apr[k])-rhs(apr[k])),k,1,n1),
      v1:listofvars(L),
      sist:makelist(diff(L,v1[k])=0,k,1,length(v1)),
      s:solve(sist,v1),
      sv:makelist(ev(f,s[k]),k,1,length(s)),
      m:lmin(sv),
      spr:sublist(s,lambda([x],ev(f,x)=m)),
      [m,makelist(subst(spr[k],v),k,1,length(spr))]
      )$
```

```
(%i2) maximize_lagr(f,apr):=block([n,n1,L,l,v,v1,M,sist,s,sv,spr],
      v:listofvars([f,apr]),
      n:length(v),
      n1:length(apr),
      L:f+sum(l[k]*(lhs(apr[k])-rhs(apr[k])),k,1,n1),
      v1:listofvars(L),
      sist:makelist(diff(L,v1[k])=0,k,1,length(v1)),
      s:solve(sist,v1),
      sv:makelist(ev(f,s[k]),k,1,length(s)),
      M:lmax(sv),
      spr:sublist(s,lambda([x],ev(f,x)=M)),
      [M,makelist(subst(spr[k],v),k,1,length(spr))]
      )$
```

Jei sąlyginiai lokalieji ekstemumai nesutampa su absoliučiais ekstemumais, tai šios komandos jų neranda.
Pabandykite jas modifikuoti sąlyginių lokaliųjų ekstemumų suradimui.

1 pvz

[4], p. 115

```
(%i3) f:x-2*y+2*z;
```
(%o3) $2z-2y+x$

```
(%i4) apr:[x^2+y^2+z^2=1];
```
(%o4) $[z^2+y^2+x^2=1]$

```
(%i5) minimize_lagr(f,apr);
```
(%o5) $[-3,[[-\frac{1}{3},\frac{2}{3},-\frac{2}{3}]]]$

```
(%i6) maximize_lagr(f,apr);
```
(%o6) $[3,[[\frac{1}{3},-\frac{2}{3},\frac{2}{3}]]]$

2 pvz

[1] , p. 971, example 5

```
(%i7) f:x+2*y+3*z;
```
(%o7) $3z+2y+x$

```
(%i8) apr:[x-y+z=1, x^2+y^2=1];
```
(%o8) $[z-y+x=1,y^2+x^2=1]$

```
(%i9) minimize_lagr(f,apr),expand;
```
(%o9) $[3-\sqrt{29},[[\frac{2}{\sqrt{29}},-\frac{5}{\sqrt{29}},1-\frac{7}{\sqrt{29}}]]]$

```
(%i10) maximize_lagr(f,apr),expand;
```
(%o10) $[\sqrt{29}+3,[[-\frac{2}{\sqrt{29}},\frac{5}{\sqrt{29}},\frac{7}{\sqrt{29}}+1]]]$

3 pvz

[1], exercise 14.8.8

```
(%i11) f:8*x-4*z;
```
(%o11) $8x-4z$

```
(%i12) apr:[x^2+10*y^2+z^2=5];
```
(%o12) $[z^2+10y^2+x^2=5]$

(%i13) minimize_lagr(f,apr);

(%o13) $[-20,[[-2,1,0]]]$

(%i14) maximize_lagr(f,apr);

(%o14) $[20,[[2,-1,0]]]$

4 pvz

[1], exercise 14.8.9

(%i15) f:x*y*z;

(%o15) $x\,y\,z$

(%i16) apr:[x^2+2*y^2+3*z^2=6];

(%o16) $[3\,z^2+2\,y^2+x^2=6]$

(%i17) minimize_lagr(f,apr);

(%o17) $[-\frac{2}{\sqrt{3}},[[-\sqrt{2},-1,-\frac{\sqrt{2}}{\sqrt{3}}],[-\sqrt{2},1,\frac{\sqrt{2}}{\sqrt{3}}],[\sqrt{2},-1,\frac{\sqrt{2}}{\sqrt{3}}],[\sqrt{2},1,-\frac{\sqrt{2}}{\sqrt{3}}]]]$

(%i18) maximize_lagr(f,apr);

(%o18) $[\frac{2}{\sqrt{3}},[[-\sqrt{2},-1,\frac{\sqrt{2}}{\sqrt{3}}],[-\sqrt{2},1,-\frac{\sqrt{2}}{\sqrt{3}}],[\sqrt{2},-1,-\frac{\sqrt{2}}{\sqrt{3}}],[\sqrt{2},1,\frac{\sqrt{2}}{\sqrt{3}}]]]$

5 pvz

[1], exercise 14.8.10

(%i19) f:x^2*y^2*z^2;

(%o19) $x^2\,y^2\,z^2$

(%i20) apr:[x^2+y^2+z^2=1];

(%o20) $[z^2+y^2+x^2=1]$

(%i21) reset(%rnum)$

(%i22) minimize_lagr(f,apr);

(%o22) $[0,[[\%r1,\sqrt{1-\%r1^2},0],[\%r2,-\sqrt{1-\%r2^2},0],[\%r3,0,\sqrt{1-\%r3^2}],[\%r4,0,-\sqrt{1-\%r4^2}],[0,\%r5,\sqrt{1-\%r5^2}],[0,\%r6,-\sqrt{1-\%r6^2}],[-1,0,0],[1,0,0],[0,1,0],[0,-1,0],[-\frac{1}{\sqrt{2}},-\frac{1}{\sqrt{2}},0],[\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}},0],[-\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}},0],[\frac{1}{\sqrt{2}},-\frac{1}{\sqrt{2}},0],[0,0,-1],[0,0,1]]]$

Nesunku pastebėti, kad užtenka imti tik pirmuosius 6 sprendinius su r, kai |r|<=1.

(%i23) maximize_lagr(f,apr);

(%o23) $[\frac{1}{27},[[-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}]]]$

6 pvz

[1], exercise 14.8.11

(%i24) f:x^2+y^2+z^2;

(%o24) $z^2+y^2+x^2$

(%i25) apr:[x^4+y^4+z^4=1];

(%o25) $[z^4+y^4+x^4=1]$

(%i26) realonly:true;

(%o26) true

(%i27) minimize_lagr(f,apr);

(%o27) [1,[[1,0,0],[-1,0,0],[0,1,0],[0,-1,0],[0,0,1],[0,0,-1]]]

(%i28) maximize_lagr(f,apr);

(%o28) $[\sqrt{3},[[\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}}],[\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}},\frac{1}{3^{1/4}}],[-\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}}],[-\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}},\frac{1}{3^{1/4}}],[\frac{1}{3^{1/4}},\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}}],[\frac{1}{3^{1/4}},\frac{1}{3^{1/4}},\frac{1}{3^{1/4}}],[-\frac{1}{3^{1/4}},\frac{1}{3^{1/4}},-\frac{1}{3^{1/4}}],[-\frac{1}{3^{1/4}},\frac{1}{3^{1/4}},\frac{1}{3^{1/4}}]]]$

7 pvz

[1], exercise 14.8.12

(%i29) f:x^4+y^4+z^4;

(%o29) $z^4+y^4+x^4$

(%i30) apr:[x^2+y^2+z^2=1];

(%o30) $[z^2+y^2+x^2=1]$

(%i31) minimize_lagr(f,apr);

(%o31) $[\frac{1}{3},[[-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[-\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}},\frac{1}{\sqrt{3}}],[\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}},-\frac{1}{\sqrt{3}}]]]$

(%i32) maximize_lagr(f,apr);

(%o32) $[1,[[-1,0,0],[1,0,0],[0,1,0],[0,-1,0],[0,0,-1],[0,0,1]]]$

8 pvz

[1], exercise 14.8.13

(%i33) f:x+y+z+t;

(%o33) $z+y+x+t$

(%i34) apr:[x^2+y^2+z^2+t^2=1];

(%o34) $[z^2+y^2+x^2+t^2=1]$

(%i35) minimize_lagr(f,apr);

(%o35) $[-2,[[-\frac{1}{2},-\frac{1}{2},-\frac{1}{2},-\frac{1}{2}]]]$

(%i36) maximize_lagr(f,apr);

(%o36) $[2,[[\frac{1}{2},\frac{1}{2},\frac{1}{2},\frac{1}{2}]]]$

9 pvz

[1], exercise 14.8.15

(%i37) f:x+2*y;

(%o37) $2\,y+x$

(%i38) apr:[x+y+z=1, y^2+z^2=4];

(%o38) $[z+y+x=1,z^2+y^2=4]$

(%i39) minimize_lagr(f,apr);

(%o39) $[1-2^{3/2},[[1,-\sqrt{2},\sqrt{2}]]]$

(%i40) maximize_lagr(f,apr);

(%o40) $[2^{3/2}+1,[[1,\sqrt{2},-\sqrt{2}]]]$

10 pvz

[1], exercise 14.8.16

```
(%i41) f:3*x-y-3*z;
```
(%o41) $-3\,z-y+3\,x$

```
(%i42) apr:[x+y-z=0, x^2+2*z^2=1];
```
(%o42) $[-z+y+x=0,2\,z^2+x^2=1]$

```
(%i43) minimize_lagr(f,apr);
```
(%o43) $[-2^{3/2}\sqrt{3},[[-\frac{\sqrt{2}}{\sqrt{3}},\frac{\sqrt{3}}{\sqrt{2}},\frac{1}{\sqrt{2}\,\sqrt{3}}]]]$

```
(%i44) maximize_lagr(f,apr);
```
(%o44) $[2^{3/2}\sqrt{3},[[\frac{\sqrt{2}}{\sqrt{3}},-\frac{\sqrt{3}}{\sqrt{2}},-\frac{1}{\sqrt{2}\,\sqrt{3}}]]]$

11 pvz

[1], exercise 14.8.17

```
(%i45) f:y*z+x*y;
```
(%o45) $y\,z+x\,y$

```
(%i46) apr:[x*y=1, y^2+z^2=1];
```
(%o46) $[x\,y=1,z^2+y^2=1]$

```
(%i47) minimize_lagr(f,apr);
```
(%o47) $[\frac{1}{2},[[\sqrt{2},\frac{1}{\sqrt{2}},-\frac{1}{\sqrt{2}}],[-\sqrt{2},-\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}]]]$

```
(%i48) maximize_lagr(f,apr);
```
(%o48) $[\frac{3}{2},[[\sqrt{2},\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}],[-\sqrt{2},-\frac{1}{\sqrt{2}},-\frac{1}{\sqrt{2}}]]]$

12 pvz

[1], exercise 14.8.18

```
(%i49) f:2*x^2+3*y^2-4*x-5;
```
(%o49) $3\,y^2+2\,x^2-4\,x-5$

```
(%i50) apr:[ x^2+y^2=16];
```
(%o50) $[y^2+x^2=16]$

```
(%i51) minimize_lagr(f,apr);
```
(%o51) [11,[[4,0]]]

```
(%i52) maximize_lagr(f,apr);
```
(%o52) $[47,[[-2,2\sqrt{3}],[-2,-2\sqrt{3}]]]$

```
(%i53) apr1:[ x^2+y^2<=16];
```
(%o53) $[y^2+x^2<=16]$

```
(%i54) load(nopt);
```
(%o54) *C:/Users/Aleksas/maxima/nopt.mac*

```
(%i55) minimize_nopt(f,apr1);
```
(%o55) $[-7,[x=1,y=0]]$

```
(%i56) maximize_nopt(f,apr1);
```
(%o56) $[47,[x=-2,y=-2\sqrt{3}],[x=-2,y=2\sqrt{3}]]$

13 pvz

[1], exercise 14.8.19

```
(%i57) f:exp(-x*y);
```
(%o57) $\%e^{-x\,y}$

```
(%i58) apr:[x^2+4*y^2=1];
```
(%o58) $[4\,y^2+x^2=1]$

```
(%i59) minimize_lagr(f,apr);
```
(%o59) $[\%e^{-\frac{1}{4}},[[\frac{1}{\sqrt{2}},\frac{1}{2^{3/2}}],[-\frac{1}{\sqrt{2}},-\frac{1}{2^{3/2}}]]]$

```
(%i60) maximize_lagr(f,apr);
```
(%o60) $[\%e^{1/4},[[\frac{1}{\sqrt{2}},-\frac{1}{2^{3/2}}],[-\frac{1}{\sqrt{2}},\frac{1}{2^{3/2}}]]]$

14 pvz

[1], exercise 14.8.20

```
(%i61) f:x^3+y^3+3*x*y;
```
(%o61) $y^3+3\,x\,y+x^3$

```
(%i62) apr:[(x-3)^2+(y-3)^2=9];
```
(%o62) $[(y-3)^2+(x-3)^2=9]$

(%i63) minimize_lagr(f,apr);

(%o63) $\left[\frac{3(3\sqrt{2}-6)^2}{4}-\frac{(3\sqrt{2}-6)^3}{4},\left[\left[-\frac{3\sqrt{2}-6}{2},-\frac{3\sqrt{2}-6}{2}\right]\right]\right]$

(%i64) expand(%);

(%o64) $\left[\frac{351}{2}-\frac{243}{\sqrt{2}},\left[\left[3-\frac{3}{\sqrt{2}},3-\frac{3}{\sqrt{2}}\right]\right]\right]$

(%i65) float(%), numer;

(%o65) [3.673052171668957,[[0.87867965644036,0.87867965644036]]]

(%i66) maximize_lagr(f,apr);

(%o66) $\left[\frac{(3\sqrt{2}+6)^3}{4}+\frac{3(3\sqrt{2}+6)^2}{4},\left[\left[\frac{3\sqrt{2}+6}{2},\frac{3\sqrt{2}+6}{2}\right]\right]\right]$

(%i67) expand(%);

(%o67) $\left[\frac{243}{\sqrt{2}}+\frac{351}{2},\left[\left[\frac{3}{\sqrt{2}}+3,\frac{3}{\sqrt{2}}+3\right]\right]\right]$

(%i68) float(%), numer;

(%o68) [347.3269478283311,[[5.121320343559642,5.121320343559642]]]

15

Apskaičiuokime atstumą tarp prasilenkiančiųjų tiesių
(x-3)/2 = (y+1)/3 = (z-2)/1 ir (x+1)/4 = (y+5)/2 = z/(-3).

(%i69) f:(x2-x1)^2+(y2-y1)^2+(z2-z1)^2;

(%o69) $(z2-z1)^2+(y2-y1)^2+(x2-x1)^2$

(%i70) apr:[(x1-3)/2=(y1+1)/3,(y1+1)/3=(z1-2),(x2+1)/4=(y2+5)/2,(y2+5)/2=z2/

(%o70) $\left[\frac{x1-3}{2}=\frac{y1+1}{3},\frac{y1+1}{3}=z1-2,\frac{x2+1}{4}=\frac{y2+5}{2},\frac{y2+5}{2}=-\frac{z2}{3}\right]$

(%i71) minimize_lagr(f,apr);

(%o71) $\left[\frac{80}{57},\left[\left[-\frac{5}{57},-\frac{49}{57},-\frac{107}{19},-\frac{281}{57},\frac{26}{57},-\frac{2}{19}\right]\right]\right]$

Atsakymas:

(%i72) ats:sqrt(%[1]);

(%o72) $\frac{4\sqrt{5}}{\sqrt{57}}$

(%i73) float(%), numer;

(%o73) 1.184697755518185

arba

```
(%i74) 20/sqrt(285);
```
(%o74) $\frac{20}{\sqrt{285}}$

```
(%i75) float(%), numer;
(%o75) 1.184697755518185
```

16

[1] , p. 969, example 3

Ant sferos x^2+y^2+z^2 = 4, rasime taškus, arčiausiai ir toliausiai nutolusius iki taško [3, 1, -1].

```
(%i76) f:(x-3)^2+(y-1)^2+(z+1)^2;
```
(%o76) $(z+1)^2+(y-1)^2+(x-3)^2$

```
(%i77) apr:[x^2+y^2+z^2 = 4];
```
(%o77) $[z^2+y^2+x^2=4]$

```
(%i78) load (sqdnst)$
```

```
(%i79) minimize_lagr(f,apr),expand;
```
(%o79) $[15-4\sqrt{11},[[\frac{6}{\sqrt{11}},\frac{2}{\sqrt{11}},-\frac{2}{\sqrt{11}}]]]$

```
(%i80) d_min=sqrt(%[1]);
```
(%o80) $d\_min=\sqrt{15-4\sqrt{11}}$

```
(%i81) sqrtdenest(%);
```
(%o81) $d\_min=\sqrt{11}-2$

```
(%i82) maximize_lagr(f,apr),expand;
```
(%o82) $[4\sqrt{11}+15,[[-\frac{6}{\sqrt{11}},-\frac{2}{\sqrt{11}},\frac{2}{\sqrt{11}}]]]$

```
(%i83) d_max=sqrt(%[1]);
```
(%o83) $d\_max=\sqrt{4\sqrt{11}+15}$

```
(%i84) sqrtdenest(%);
```
(%o84) $d\_max=\sqrt{11}+2$

17 pvz

[3], p. 104, 6.12 pavyzdys
Rasime elipsoido paviršiaus x^2/96+y^2+z^2 = 1 taškus,
labiausiai nutolusius nuo plokštumos 3x+4y+12z = 288.

(%i85) `f:(x-x1)^2+(y-y1)^2+(z-z1)^2;`

(%o85) $(z-z1)^2+(y-y1)^2+(x-x1)^2$

(%i86) `apr:[x^2/96+y^2+z^2 = 1,3*x1+4*y1+12*z1=288];`

(%o86) $[z^2+y^2+\frac{x^2}{96}=1,12\,z1+4\,y1+3\,x1=288]$

(%i87) `spr1:minimize_lagr(f,apr);`

(%o87) $[\frac{65536}{169},[[9,\frac{2289}{169},\frac{1}{8},\frac{8361}{1352},\frac{3}{8},\frac{25083}{1352}]]]$

Sprendinių reikšmių išvedimo tvarka matomai yra: x, x1, y, y1, z, z1.

(%i88) `d_min=sqrt(%[1]);`

(%o88) $d\_min=\frac{256}{13}$

Arčiausias iki plokštumos elipsoido taškas yra

(%i89) `[x=spr1[2][1][1],y=spr1[2][1][3],z=spr1[2][1][5]];`

(%o89) $[x=9,y=\frac{1}{8},z=\frac{3}{8}]$

(%i90) `spr2:maximize_lagr(f,apr);`

(%o90) $[\frac{102400}{169},[[-9,-\frac{561}{169},-\frac{1}{8},\frac{10071}{1352},-\frac{3}{8},\frac{30213}{1352}]]]$

(%i91) `d_max=sqrt(%[1]);`

(%o91) $d\_max=\frac{320}{13}$

Toliausias iki plokštumos elipsoido taškas yra

(%i92) `[x=spr2[2][1][1],y=spr2[2][1][3],z=spr2[2][1][5]];`

(%o92) $[x=-9,y=-\frac{1}{8},z=-\frac{3}{8}]$

Literatūra:
[1] James Stewart, Calculus 5th Edition,
[2] Stewart.4th.Edition.Multivariable.Calculus.Teacher's.Edition.Solutions.Manuel.(Math.32.Series).pdf
[3] G.Stepanauskas, A.Raudeliūnas, Kelių kintamųjų funkcijos, V., VU, 1995
[4] V.Kabaila, Matematinė analizė, 2d. V. , 1986
[5] http://en.wikipedia.org/wiki/Lagrange_multiplier