TP uždavinio sprendimo pavyzdys.

A.Domarkas, VU, 2016

Pavyzdys iš A.Žilinskas, Matematinis programavimas, 2005, psl. 128-143
Raskite minimumą funkcijos f = x1+2*x2+3*x3+4*x4,
kai 2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4, x1>=0, x2>=0, x3>=0, x4>=0.

Figure 1:

$$\min\left(x_1+2x_2+3x_3+4x_4\right),$$
$$\begin{aligned} 2x_1+x_2+5x_3+2x_4 &= 10,\\ x_1+x_2-x_3-x_4 &= 4,\\ x_i &\geq 0 \end{aligned}$$

1 būdas

```
(%i1) load(simplex)$
```

```
(%i2) f:x1+2*x2+3*x3+4*x4;
(%o2) 4 x4+3 x3+2 x2+x1
```

```
(%i3) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4];
(%o3) [2 x4+5 x3+x2+2 x1=10,-x4-x3+x2+x1=4]
```

```
(%i4) minimize_lp(f,apr),nonegative_lp=true;
```
(%o4) $[\frac{36}{7},[x4=0,x3=\frac{2}{7},x2=0,x1=\frac{30}{7}]]$

2 būdas

```
(%i5) A:matrix([2,1,5,2],[1,1,-1,-1]);
```
(%o5) $\begin{bmatrix} 2 & 1 & 5 & 2 \\ 1 & 1 & -1 & -1 \end{bmatrix}$

```
(%i6) c:[1,2,3,4];
(%o6) [1,2,3,4]
```

```
(%i7) b:[10,4];
(%o7) [10,4]
```

```
(%i8) load(simplex)$
```

```
(%i9) linear_program(A, b, c);
```
(%o9) $[[\frac{30}{7},0,\frac{2}{7},0],\frac{36}{7}]$

3 būdas. Pasinaudosime programa "nopt"((C), A.Domarkas)
netiesinės optimizacijos uždavinių sprendimui.

```
(%i10) load(nopt);
(%o10) C:/Users/aleksas/maxima/nopt.mac
```

```
(%i11) f:x1+2*x2+3*x3+4*x4;
(%o11) 4 x4+3 x3+2 x2+x1
```

```
(%i12) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4,x1>=0, x2>=0, x3>=0, x4>=0];
(%o12) [2 x4+5 x3+x2+2 x1=10,-x4-x3+x2+x1=4,x1>=0,x2>=0,x3>=0,x4>=0]
```

(%i13) minimize_nopt(f,apr);

(%o13) $[\frac{36}{7},[x1=\frac{30}{7},x2=0,x3=\frac{2}{7},x4=0]]$

4 būdas. Randame lygčių sistemos bazinius sprendinius.
Žr. [1] 244-252; [5], 127-134

Apibrėžimas. Tarkime, kad lygčių sistema yra suderinta ir matricos rangas r yra mažesnis negu nežinomųjų sksičius n. Tada jos sprendinys, kurio laisvųjų nežinomųjų reikšmės lygios nuliui, vadinamas baziniu sprendiniu.

Pastaba. Laisvųjų nežinomųjųjų skaičius yra k = n - r.

(%i14) f:x1+2*x2+3*x3+4*x4;

(%o14) $4\,x4+3\,x3+2\,x2+x1$

(%i15) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4,x1>=0, x2>=0, x3>=0, x4>=0];

(%o15) $[2\,x4+5\,x3+x2+2\,x1=10,-x4-x3+x2+x1=4,x1>=0,x2>=0,x3>=0,x4>=0]$

(%i16) eq1:apr[1];

(%o16) $2\,x4+5\,x3+x2+2\,x1=10$

(%i17) eq2:apr[2];

(%o17) $-x4-x3+x2+x1=4$

(%i18) sist:[eq1,eq2]$

(%i19) vars:[x1,x2,x3,x4]$

(%i20) var:full_listify(powerset({x1,x2,x3,x4}, 2));

(%o20) $[[x1,x2],[x1,x3],[x1,x4],[x2,x3],[x2,x4],[x3,x4]]$

(%i21) makelist(solve(append(sist,var[i]),vars),i,1,length(var));

(%o21) $[[[x1=0,x2=0,x3=6,x4=-10]],[[x1=0,x2=6,x3=0,x4=2]],[[x1=0,x2=5,x3=1,x4=0]],[[x1=\frac{9}{2},x2=0,x3=0,x4=\frac{1}{2}]],[[x1=\frac{30}{7},x2=0,x3=\frac{2}{7},x4=0]],[[x1=6,x2=-2,x3=0,x4=0]]]$

(%i22) delete([],%)$

Radome visus bazinius sprendinius:

(%i23) baz:makelist(subst(%[i],vars),i,1,length(%));

(%o23) $[[0,0,6,-10],[0,6,0,2],[0,5,1,0],[\frac{9}{2},0,0,\frac{1}{2}],[\frac{30}{7},0,\frac{2}{7},0],[6,-2,0,0]]$

(%i24) length(%);

(%o24) 6

Apribojimų aibės kraštutiniai taškai yra baziniai sprendiniai
su neneigiamomis koordinatėmis:

(%i25) T:sublist(baz,lambda([e],lmin(e)>=0));

(%o25) $[[0,6,0,2],[0,5,1,0],[\frac{9}{2},0,0,\frac{1}{2}],[\frac{30}{7},0,\frac{2}{7},0]]$

Apskaičiuosime tikslo funkcijos reikšmes šiuose taškuose.

(%i26) r:makelist(ev(f,[x1=T[k][1],x2=T[k][2],x3=T[k][3],x4=T[k][4]]),
k,1,length(T));

(%o26) $[20,13,\frac{13}{2},\frac{36}{7}]$

Mažiausioji reikšmė yra:

```
(%i27) m:lmin(%);
```

(%o27) $\frac{36}{7}$

Ji įgyjama taškuose:

```
(%i28) sublist_indices (r,lambda([x],x=m))$
       min_taskai:makelist(T[k],k,%);
```

(%o29) $[[\frac{30}{7},0,\frac{2}{7},0]]$

Atsakymas: f_min=36/7, kai [x1,x2,x3,x4]=[30/7,0,2/7,0].

5 būdas. Sprendimas, randant kraštutinius taškus.

Pasinaudosime programa "ext"((C), A.Domarkas)
kraštutinių taškų(extreme points) radimui.

Tikslo funkcijos minimumas pasiekiamas šiuose taškuose.
Toliau toliau iš jų atrenkame tuos, kuriuose pasiekiamas minimumas.

```
(%i30) kill(all)$
```

```
(%i1) load(simplex)$
```

```
(%i2) ext(apr):=block([var,fs,cs,ap,s,S,m],
      var:sort(listofvars(apr)),
      s:apply("+",var),
      fs:append([1,s,-s],var,-var),
      ap(k):=subst(apr[k]=(lhs(apr[k])=rhs(apr[k])),apr),
      cs:makelist(ap(k),k,1,length(apr)),
      S:[],
      for f in fs do
      for c in cs do
      (
      m:minimize_lp(f,c),
      if listp(m) then
      S:cons(subst(m[2],var),S)
      ),
      listify(setify(S))
      )$
```

```
(%i3) f:x1+2*x2+3*x3+4*x4;
```

(%o3) $4\,x4+3\,x3+2\,x2+x1$

```
(%i4) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4,x1>=0, x2>=0, x3>=0, x4>=0];
```

(%o4) $[2\,x4+5\,x3+x2+2\,x1=10,-x4-x3+x2+x1=4,x1>=0,x2>=0,x3>=0,x4>=0]$

```
(%i5) T:ext(apr);
```

(%o5) $[[0,5,1,0],[0,6,0,2],[\frac{30}{7},0,\frac{2}{7},0],[\frac{9}{2},0,0,\frac{1}{2}]]$

Apskaičiuosime tikslo funkcijos reikšmes šiuose taškuose.

```
(%i6) r:makelist(ev(f,
      [x1=T[k][1],x2=T[k][2],x3=T[k][3],x4=T[k][4]]),k,1,length(T));
```

(%o6) $[13,20,\frac{36}{7},\frac{13}{2}]$

Mažiausioji reikšmė yra:

```
(%i7) m:lmin(%);
```

(%o7) $\frac{36}{7}$

Ji įgyjama taškuose:

```
(%i8) sublist_indices (r,lambda([x],x=m))$
      min_taskai:makelist(T[k],k,%);
```

(%o9) $[[\frac{30}{7}, 0, \frac{2}{7}, 0]]$

Atsakymas: f_min=36/7, kai [x1,x2,x3,x4]=[30/7,0,2/7,0].

6 būdas su COBYLA

```
(%i10) f:x1+2*x2+3*x3+4*x4;
```

(%o10) $4\, x4+3\, x3+2\, x2+x1$

```
(%i11) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4,x1>=0, x2>=0, x3>=0, x4>=0];
```

(%o11) $[2\, x4+5\, x3+x2+2\, x1=10, -x4-x3+x2+x1=4, x1>=0, x2>=0, x3>=0, x4>=0]$

```
(%i44) load(fmin_cobyla)$
```

```
(%i13) fmin_cobyla(f,[x1,x2,x3,x4],[0,0,0,0], constraints = apr, iprint=1);
  Normal return from subroutine COBYLA
  NFVALS =    46    F = 5.142857E+00    MAXCV = 4.469613E-16
  X = 4.285714E+00   0.000000E+00   2.857143E-01  -1.436060E-18
```

(%o13) $[[x1=4.285714285714286, x2=0.0, x3=0.2857142857142857, x4=-1.436059937033911\; 10^{-18}], 5.142857142857142, 46, 0]$

Anksčiau su "simplex" paketu buvo gauta

```
(%i14) minimize_lp(f,apr),nonegative_lp=true;
```

(%o14) $[\frac{36}{7}, [x4=0, x3=\frac{2}{7}, x2=0, x1=\frac{30}{7}]]$

```
(%i15) float(%);
```

(%o15) $[5.142857142857143, [x4=0.0, x3=0.2857142857142857, x2=0.0, x1=4.285714285714286]]$

Matome, kad tai sutampa su COBYLA rezultatais.

7 būdas. Suvedant į dvimatį uždavinį, kurį sprendžiame grafiniu metodu.
Su animacija parodome tikslo funkcijos lygio linijos judėjimą
gradiento kryptimi.

```
(%i16) f:x1+2*x2+3*x3+4*x4;
```

(%o16) $4\, x4+3\, x3+2\, x2+x1$

```
(%i17) apr:[2*x1+x2+5*x3+2*x4=10,x1+x2-x3-x4=4,x1>=0, x2>=0, x3>=0, x4>=0];
```

(%o17) $[2\, x4+5\, x3+x2+2\, x1=10, -x4-x3+x2+x1=4, x1>=0, x2>=0, x3>=0, x4>=0]$

Eliminuosime kintamuosius x3 ir x4:

```
(%i18) s34:solve([apr[1],apr[2]],[x3,x4]);
```

(%o18) $[[x3=-\frac{3\, x2+4\, x1-18}{3}, x4=\frac{6\, x2+7\, x1-30}{3}]]$

```
(%i19) apr1:subst(%[1],apr),expand;
```

(%o19) $[10=10, 4=4, x1>=0, x2>=0, -x2-\frac{4\, x1}{3}+6>=0, 2\, x2+\frac{7\, x1}{3}-10>=0]$

```
(%i20) f1:subst(s34,f),expand;
```

(%o20) $7\,x2+\frac{19\,x1}{3}-22$

```
(%i42) load(draw)$
       ratprint:false$
```

```
(%i23) set_draw_defaults(
                  x_voxel = 40,
                  y_voxel = 40,
                  xrange = [0,6],
                  yrange = [0,6],
                  xlabel = "x1",
                  ylabel = "x2",
                  grid   = true,
                  proportional_axes = xy,
               fill_color = skyblue)$
```

```
(%i24) sritis:apply("and", apr1);
```

(%o24) $x1>=0$ and $x2>=0$ and $-x2-\frac{4\,x1}{3}+6>=0$ and $2\,x2+\frac{7\,x1}{3}-10>=0$

Brėžinys su animacija:

```
(%i25) with_slider_draw(
                 z, makelist(i, i, 3, 8),
                 region(sritis, x1, 0, 6, x2, 0, 6),
                 key   = string(ev(f1,nouns)=z),
                 implicit(f1=z,x1,0,6, x2, 0, 6),
                 color=red,
                 key   = string(-x2-(4*x1)/3+6=0),
                 implicit(-x2-(4*x1)/3+6=0,x1,0,6,x2, 0, 6),
                 color=green,
                 key   = string(2*x2+(7*x1)/3-10=0),
                 implicit(2*x2+(7*x1)/3-10=0,x1,0,6,x2, 0, 6)
       )$
```

(%t25)



Matome, kad minimumas yra pasiekiamas tiesių x2=0 ir 2*x2+(7*x1)/3-10=0 susikirtimo taške.

```
(%i26) sol12:solve([x2=0, 2*x2+(7*x1)/3-10=0]);
```

(%o26) $[\,[x1=\frac{30}{7},x2=0]\,]$

```
(%i27) sol34:subst(sol12,s34);
```

(%o27) $[[x3=\frac{2}{7}, x4=0]]$

Atsakymas:

```
(%i28) flatten([sol12,sol34]);
```

(%o28) $[x1=\frac{30}{7}, x2=0, x3=\frac{2}{7}, x4=0]$

```
(%i29) f_min=subst(%,f);
```

(%o29) $f\_min=\frac{36}{7}$

8 būdas. Sudarant ir išsprendžiant dualųjį uždavinį.
Bendrojo TP uždavinio dualaus uždavinio formulavimą žr. [2], 83 psl.; [6]

Komanda "dual" randa bendrojo minimizavimo uždavinio dualųjį.

```
(%i30) dual(A,b,c,I0,J0):=block([m,n,I,J,JminusJ0,Y],
       m:matrix_size(A)[1],
       n:matrix_size(A)[2],
       I:makelist(k,k,1,m),
       J:makelist(k,k,1,n),
       JminusJ0:sublist(J,lambda([x],not member(x,J0))),
       Y:makelist(concat(y,k),k,I),
       [b.Y,append(
       makelist((transpose(A).Y)[j,1]<=c[j],j,J0),
       makelist((transpose(A).Y)[j,1]=c[j],j,JminusJ0),
       makelist(Y[i]>=0,i,I0))]
       )$
```

Čia
A - koeficientų matrica,
b - dešiniųjų pusių vektorius,
c - tikslo funkcijos koeficientų vektorius,
I0 - indeksų i sąrašas, su kuriais yra nelygybė ">=",
J0 - indeksų j sąrašas, su kuriais xj>=0.

Pastaba. Jei yra nelygybių su "<=" ženklu, tai pradžioje
jas reikia padauginti iš -1. Minimizavimo uždavinyje
visos nelygybės turi būti su ">=" ženklu. Maksimizavimo
uždavinyje visos nelygybės turi būti su "<=" ženklu.
Komanda "dual" yra skirta bendram minimizavimo uždaviniui.
Patys sudarykite analogišką komandą maksimizavimo uždaviniui.
Žr. [6]

```
(%i31) A:matrix([2,1,5,2],[1,1,-1,-1]);
```

(%o31) $\begin{bmatrix} 2 & 1 & 5 & 2 \\ 1 & 1 & -1 & -1 \end{bmatrix}$

```
(%i32) b:[10,4];
```

(%o32) $[10, 4]$

```
(%i33) c:[1,2,3,4];
```

(%o33) $[1, 2, 3, 4]$

```
(%i34) I0:[];
```

(%o34) $[]$

```
(%i35) J0:[1,2,3,4];
```

(%o35) $[1, 2, 3, 4]$

Dualusis uždavinys:

```
(%i37) dp:dual(A,b,c,I0,J0);
```

(%o37) $[4\, y2+10\, y1, [y2+2\, y1<=1, y2+y1<=2, 5\, y1-y2<=3, 2\, y1-y2<=4]]$

```
(%i38) f1:dp[1];
```

(%o38) $4\, y2+10\, y1$

```
(%i39) apr1:dp[2];
```

(%o39) $[y2+2\, y1<=1, y2+y1<=2, 5\, y1-y2<=3, 2\, y1-y2<=4]$

```
(%i40) load(simplex)$
```

```
(%i41) maximize_lp(f1,apr1);
```

(%o41) $[\frac{36}{7}, [y2=-\frac{1}{7}, y1=\frac{4}{7}]]$

Atsakymas:
f_min = f1_max = 36/7

Palyginkite šiuos sprendimo būdus su [5], psl. 128-143

[1] A. Apynis, E. Stankus, Matematikos pagrindai, V., TEV, 2009
[2] A. Apynis, Optimizavimo metodai, V., VU, 2005
[3] V. Pekarskas, A.Pekarskienė, Tiesinės algebros ir analizinės geometrijos elementai, 2004
[4] V. Čiočys, R.Jasiulionis, Matematinis programavimas, V., Mokslas, 1990
[5] A.Žilinskas, Matematinis programavimas, 2005
[6] https://www.cs.cmu.edu/afs/cs.cmu.edu/academic/class/15859-f11/www/notes/lecture05.pdf