Baziniai sprendiniai ir kraštutiniai taškai

(C) A.Domarkas, VU, 2015

žr.: [1] 244-252; [2] 194-198; [3] 33-40; [4] 89-98; [5] 6.3

Tegul tiesinių lygčių sistemos nežinomųjų skaičius yra n
ir koeficientų matricos A rangas lygus r.

Tiesinių lygčių sistemos nežinomieji, kurių reikšmes galima laisvai pasirinkti, vadinami laisvaisiais nežinomaisiais. Kiti šios sistemos nežinomieji vadinami baziniais nežinomaisiais. Bazinių nežinomųjų skaičius yra lygus matricos A rangui r, o laisvųjų nežinomųjų skaičius yra lygus k = n-r.

Apibrėžimas. Tarkime, kad lygčių sistema yra suderinta ir r < n. Tada jos sprendinys, kurio laisvųjų nežinomųjų reikšmės lygios nuliui, vadinamas baziniu sprendiniu.

Baziniai sprendiai yra pritaikomi tiesinio programavimo uždaviniuose.
Jie yra svarbūs apbūdinant leistinosios aibės kraštutinius taškus.
Ekstremalios tikslo funkcijos reikšmės yra įgyjamos šiuose taškuose.

Apibrėžimas. Tegu X yra iškiloji aibė. Taškas x0 yra vadinamas aibės X kraštutiniu tašku(extreme point), jei jis nepriklauso vidui atkarpos, jungiančios du skirtingus taškus iš aibės X. Jei aibė X yra iškilasis daugiakampis arba iškilasis briaunainis, tai kraštutiniai taškai sutampa su jo viršūnėmis.
žr.: [4], 89-98
http://en.wikipedia.org/wiki/Extreme_point
http://ljk.imag.fr/membres/Anatoli.Iouditski/cours/convex/chapitre_1.pdf

Čia naudojama atvirojo kodo kompiuterinės algebros sistema Maxima 5.36.1

1 pavyzdys

1 pavyzdys([1], 247 p.) Raskite tiesinių lygčių sistemos visus bazinius sprendinius.

```
(%i1) eq1:x1+x2-x3+x4+x5=1$
      eq2:x2+x3+x4-x5=2$
      eq3:x1-x3+x5=0$
```

```
(%i4) sist:[eq1,eq2,eq3]$
```

```
(%i5) vars:sort(listofvars(sist));
```
(%o5) $[x1, x2, x3, x4, x5]$

```
(%i6) Ai:augcoefmatrix(sist,vars);
```
(%o6) $\begin{bmatrix} 1 & 1 & -1 & 1 & 1 & -1 \\ 0 & 1 & 1 & 1 & -1 & -2 \\ 1 & 0 & -1 & 0 & 1 & 0 \end{bmatrix}$

```
(%i7) A:submatrix(Ai,6);
```
(%o7) $\begin{bmatrix} 1 & 1 & -1 & 1 & 1 \\ 0 & 1 & 1 & 1 & -1 \\ 1 & 0 & -1 & 0 & 1 \end{bmatrix}$

```
(%i8) [m,n]:matrix_size(A);
```
(%o8) $[3, 5]$

Matricos A rangas:

```
(%i9) r:rank(A);
```
(%o9) 3

Laivųjų nežinomųjų skaičius:

```
(%i10) k:n-r;
(%o10) 2
```

```
(%i11) rank(Ai)=rank(A);
(%o11) 3=3
```

```
(%i12) is(%);
(%o12) true
```

Gavome, kad išplėstinės matricos rangas lygus koeficientų matricos rangui.
Todėl pagal Kronekerio-Kapelio teoremą sistema yra suderinta.
Didžiausias laisvųjų nežinomųjų pasirinkimo būdų skaičius
lygus derininių skaičiui iš n po k:

```
(%i13) binomial(n,k);
(%o13) 10
```

```
(%i14) var:full_listify(powerset({x1,x2,x3,x4,x5}, k));
(%o14) [[x1,x2],[x1,x3],[x1,x4],[x1,x5],[x2,x3],[x2,x4],[x2,x5],[x3,x4],[x3,x5]
,[x4,x5]]
```

```
(%i15) length(var);
(%o15) 10
```

```
(%i16) makelist(solve(append(sist,var[i]),vars),i,1,length(var));
(%o16) [[],[],[],[],[[x1=1,x2=0,x3=0,x4=1,x5=-1]],[],[[x1=1,x2=0,x3=1,x4=1,
x5=0]],[[x1=1,x2=1,x3=0,x4=0,x5=-1]],[],[[x1=1,x2=1,x3=1,x4=0,x5=0]]]
```

```
(%i17) delete([],%);
(%o17) [[[x1=1,x2=0,x3=0,x4=1,x5=-1]],[[x1=1,x2=0,x3=1,x4=1,x5=0]],[[x1=1,x2
=1,x3=0,x4=0,x5=-1]],[[x1=1,x2=1,x3=1,x4=0,x5=0]]]
```

Todėl baziniai sprendiniai yra:

```
(%i18) ats:makelist(subst(%[i],vars),i,1,length(%));
(%o18) [[1,0,0,1,-1],[1,0,1,1,0],[1,1,0,0,-1],[1,1,1,0,0]]
```

arba

```
(%i19) reverse(%);
(%o19) [[1,1,1,0,0],[1,1,0,0,-1],[1,0,1,1,0],[1,0,0,1,-1]]
```

2 pavyzdys

2 pavyzdys([3], 40 p., 11 užd.) Raskite tiesinių lygčių sistemos visus bazinius
sprendinius. Šiame pavyzdyje matricos A rangas yra mažesnis už lygčių skaičių.

```
(%i20) kill(all)$
```

```
(%i1) eq1:x1-2*x2+x4=-3$
      eq2:5*x1-4*x3-x4=5$
      eq3:-7*x1+4*x2+4*x3-x4=1$
```

```
(%i4) sist:[eq1,eq2,eq3]$
```

```
(%i5) sist:[eq1,eq2,eq3]$
```

```
(%i6) vars:sort(listofvars(sist));
(%o6) [x1,x2,x3,x4]
```

```
(%i7) Ai:augcoefmatrix(sist,vars);
```

(%o7) $\begin{bmatrix} 1 & -2 & 0 & 1 & 3 \\ 5 & 0 & -4 & -1 & -5 \\ -7 & 4 & 4 & -1 & -1 \end{bmatrix}$

```
(%i8) A:submatrix(Ai,5);
```

(%o8) $\begin{bmatrix} 1 & -2 & 0 & 1 \\ 5 & 0 & -4 & -1 \\ -7 & 4 & 4 & -1 \end{bmatrix}$

```
(%i9) [m,n]:matrix_size(A);
```

(%o9) $[3,4]$

Matricos A rangas:

```
(%i10) r:rank(A);
```

(%o10) 2

Laivųjų nežinomųjų skaičius:

```
(%i11) k:n-r;
```

(%o11) 2

```
(%i12) rank(Ai)=rank(A);
```

(%o12) $2=2$

```
(%i13) is(%);
```

(%o13) true

Gavome, kad išplėstinės matricos rangas lygus koeficientų matricos rangui.
Todėl pagal Kronekerio-Kapelio teoremą sistema yra suderinta.
Didžiausias laisvųjų nežinomųjų pasirinkimo skaičius
lygus derininių skaičiui iš n po k:

```
(%i14) binomial(n,k);
```

(%o14) 6

```
(%i15) var:full_listify(powerset({x1,x2,x3,x4}, k));
```

(%o15) $[[x1,x2],[x1,x3],[x1,x4],[x2,x3],[x2,x4],[x3,x4]]$

```
(%i16) length(var);
```

(%o16) 6

```
(%i17) makelist(solve(append(sist,var[i]),vars),i,1,length(var));
solve: dependent equations eliminated: (2)
solve: dependent equations eliminated: (3)
solve: dependent equations eliminated: (2)
solve: dependent equations eliminated: (2)
solve: dependent equations eliminated: (2)
solve: dependent equations eliminated: (3)
```

(%o17) $[[[x1=0,x2=0,x3=-\frac{1}{2},x4=-3]],[[x1=0,x2=-1,x3=0,x4=-5]],[[x1=0,x2=\frac{3}{2},x3=-\frac{5}{4},x4=0]],[[x1=\frac{1}{3},x2=0,x3=0,x4=-\frac{10}{3}]],[[x1=-3,x2=0,x3=-5,x4=0]],[[x1=1,x2=2,x3=0,x4=0]]]$

Todėl baziniai sprendiniai yra:

(%i18) `ats:makelist(subst(%[i],vars),i,1,length(%));`

(%o18) $[[0,0,-\frac{1}{2},-3],[0,-1,0,-5],[0,\frac{3}{2},-\frac{5}{4},0],[\frac{1}{3},0,0,-\frac{10}{3}],[-3,0,-5,0],[1,2,0,0]]$

arba

(%i19) `reverse(%);`

(%o19) $[[1,2,0,0],[-3,0,-5,0],[\frac{1}{3},0,0,-\frac{10}{3}],[0,\frac{3}{2},-\frac{5}{4},0],[0,-1,0,-5],[0,0,-\frac{1}{2},-3]]$

Gavome tokį pat atsakymą kaip ir [4].

3 pavyzdys

3 pavyzdys([3], 133 p.) Raskite tiesinių lygčių sistemos

(%i20)
```
eq1:x1-3*x2+2*x4-4*x5=-7$
eq2:3*x1+x2-2*x3+x5=1$
eq3:2*x1-x2+x3+7*x4-3*x5=5$
```

neneigimų sprendinių aibės X kraštutinius taškus.

(%i23) `sist:[eq1,eq2,eq3]$`

(%i24) `vars:[x1,x2,x3,x4,x5]$`

(%i25) `var:full_listify(powerset({x1,x2,x3,x4,x5}, 2));`

(%o25) $[[x1,x2],[x1,x3],[x1,x4],[x1,x5],[x2,x3],[x2,x4],[x2,x5],[x3,x4],[x3,x5],[x4,x5]]$

(%i26) `makelist(solve(append(sist,var[i]),vars),i,1,length(var));`

(%o26) $[[[x1=0,x2=0,x3=\frac{37}{46},x4=\frac{79}{46},x5=\frac{60}{23}]],[[x1=0,x2=-\frac{37}{3},x3=0,x4=\frac{14}{3},x5=\frac{40}{3}]],[[x1=0,x2=\frac{79}{11},x3=\frac{14}{11},x4=0,x5=-\frac{40}{11}]],[[x1=0,x2=3,x3=1,x4=1,x5=0]],[[x1=-\frac{37}{69},x2=0,x3=0,x4=\frac{137}{69},x5=\frac{60}{23}]],[[x1=\frac{79}{23},x2=0,x3=\frac{137}{23},x4=0,x5=\frac{60}{23}]],[],[[x1=-\frac{14}{15},x2=\frac{137}{15},x3=0,x4=0,x5=-\frac{16}{3}]],[[x1=-\frac{2}{3},x2=3,x3=0,x4=\frac{4}{3},x5=0]],[[x1=2,x2=3,x3=4,x4=0,x5=0]]]$

(%i27) `delete([],%)$`

Radome visus bazinius sprendinius:

(%i28) `baz:makelist(subst(%[i],vars),i,1,length(%));`

(%o28) $[[0,0,\frac{37}{46},\frac{79}{46},\frac{60}{23}],[0,-\frac{37}{3},0,\frac{14}{3},\frac{40}{3}],[0,\frac{79}{11},\frac{14}{11},0,-\frac{40}{11}],[0,3,1,1,0],[-\frac{37}{69},0,0,\frac{137}{69},\frac{60}{23}],[\frac{79}{23},0,\frac{137}{23},0,\frac{60}{23}],[-\frac{14}{15},\frac{137}{15},0,0,-\frac{16}{3}],[-\frac{2}{3},3,0,\frac{4}{3},0],[2,3,4,0,0]]$

(%i29) `length(%);`

(%o29) 9

Gauname atsakymą: neneigiamų sprendinių aibės X kraštutiniai taškai yra

(%i30) `krast:sublist(baz,lambda([e],lmin(e)>=0));`

(%o30) $[[0,0,\frac{37}{46},\frac{79}{46},\frac{60}{23}],[0,3,1,1,0],[\frac{79}{23},0,\frac{137}{23},0,\frac{60}{23}],[2,3,4,0,0]]$

Radome 9 bazinius sprendinius, iš kurių 4 yra neneigiami.
Vadovėlyje [3] yra rasti tik 3 kraštutiniai taškai [0,3,1,1,0], [2,3,4,0,0] ir [0,0,37/46,79/46,60/23]. Palyginkite sprendimo metodus.

4 pavyzdys

4 pavyzdys([1], 294 p.) Raskite minimumą funkcijos F=4*x1+x2-x3+2*x4-x5,
kai 3*x1+x2+2*x3-x4=5, 2*x1-x3+3*x4+4*x5=8, x1>=0, ..., x5>=0.

```
(%i31) remvalue (all)$
```

Čia yra pateikiamas kitas bazinių sprendinių ieškojimo būdas, kuris naudoja komandą "Msolve".

```
(%i32) Msolve(A,K):=block([X,S,n,m,i,j,k],
       X:makelist(x[i],i,K),
       n:length(A[1])-1, m:length(A),
       S:makelist(sum(A[i,j]*x[j],j,1,n)=A[i,n+1],i,1,m),
       if rank(A)#rank(submatrix(A,n+1)) then return("sprendinių nėra"),
       solve(S,X), if %%=[] then
       return("neteisingai baziniai pasirinkti nežinomieji") else
       spr:subst(%%,makelist(x[i],i,n)),
       p:listofvars(spr),k:length(p),
       L:[a,b,c,d,e,f,g,h],
       subst(makelist(p[i]=L[i],i,1,k),spr)
       )$
```

```
(%i33) Ai:matrix([3,1,2,-1,0,5],
                 [2,0,-1,3,4,8]);
```

(%o33) $\begin{bmatrix} 3 & 1 & 2 & -1 & 0 & 5 \\ 2 & 0 & -1 & 3 & 4 & 8 \end{bmatrix}$

```
(%i34) A:submatrix(Ai,6)$ B:col(Ai,6)$
```

```
(%i36) binomial(5,2);
```

(%o36) 10

```
(%i37) var:full_listify(powerset({1,2,3,4,5}, 2));
```

(%o37) [[1,2],[1,3],[1,4],[1,5],[2,3],[2,4],[2,5],[3,4],[3,5],[4,5]]

```
(%i38) visi:[]$
       for k thru 10 do (Msolve(Ai,var[k]),
       if listp(%%) then visi:endcons(%%,visi))$
```

Visi baziniai spendiniai:

```
(%i40) baz:visi,a=0,b=0,c=0;
```

(%o40) $[[4,-7,0,0,0],[3,0,-2,0,0],[\frac{23}{11},0,0,\frac{14}{11},0],[\frac{5}{3},0,0,0,\frac{7}{6}],[0,21,-8,0,0],[0,\frac{23}{3},0,\frac{8}{3},0],[0,5,0,0,2],[0,0,\frac{23}{5},\frac{21}{5},0],[0,0,\frac{5}{2},0,\frac{21}{8}],[0,0,0,-5,\frac{23}{4}]]$

Apibrėžiame tikslo funkciją F(x). Pabrėšime, kad čia x yra vektorius.
Kai rašome x[i], tai imama to vektoriaus i-toji koordinatė.

```
(%i41) F(x):=4*x[1]+x[2]-x[3]+2*x[4]-x[5];
```

(%o41) $F(x):=4\,x_1+x_2-x_3+2\,x_4-x_5$

Visi baziniai spendiniai su neneigiamomis koordinatėmis:

```
(%i42) baz1:sublist(baz,lambda([e],lmin(e)>=0));
```

(%o42) $[[\frac{23}{11},0,0,\frac{14}{11},0],[\frac{5}{3},0,0,0,\frac{7}{6}],[0,\frac{23}{3},0,\frac{8}{3},0],[0,5,0,0,2],[0,0,\frac{23}{5},\frac{21}{5},0],[0,0,\frac{5}{2},0,\frac{21}{8}]]$

Atitinkamos tikslo funkcijos F reikšmės yra:

(%i43) map(F,baz1);

(%o43) $[\frac{120}{11}, \frac{11}{2}, 13, 3, \frac{19}{5}, -\frac{41}{8}]$

(%i44) lmin(%);

(%o44) $-\frac{41}{8}$

Matome, kad mažiausia reikšmė yra -41/8, įgyjama ant bazinio sprendinio [0,0,5/2,0,21/8].
Bazinius sprendinius surūšiuojame taip, kad mažiausia reikšmė būtų ant pirmojo bazinio sprendin

(%i45) ru(a,b):=F(a)<=F(b)$

(%i46) sort(baz1,ru);

(%o46) $[[0,0,\frac{5}{2},0,\frac{21}{8}],[0,5,0,0,2],[0,0,\frac{23}{5},\frac{21}{5},0],[\frac{5}{3},0,0,0,\frac{7}{6}],[\frac{23}{11},0,0,\frac{14}{11},0],[0,\frac{23}{3},0,\frac{8}{3},0]]$

(%i47) ats:first(%);

(%o47) $[0,0,\frac{5}{2},0,\frac{21}{8}]$

(%i48) F_min=F(%);

(%o48) $F\_min=-\frac{41}{8}$

2 būdas(naudojant paketą "simplex"):

(%i49) load(simplex)$

(%i50) F:4*x1+x2-x3+2*x4-x5;

(%o50) $-x5+2\,x4-x3+x2+4\,x1$

(%i51) apr:[3*x1+x2+2*x3-x4=5,2*x1-x3+3*x4+4*x5=8];

(%o51) $[-x4+2\,x3+x2+3\,x1=5, 4\,x5+3\,x4-x3+2\,x1=8]$

(%i52) minimize_lp(F,apr), nonegative_lp=true;

(%o52) $[-\frac{41}{8},[x5=\frac{21}{8}, x4=0, x3=\frac{5}{2}, x2=0, x1=0]]$

3 būdas:

(%i53) kill(all)$

(%i1) eq1:3*x[1]+x[2]+2*x[3]-x[4]=5$
eq2:2*x[1]-x[3]+3*x[4]+4*x[5]=8$

(%i3) F(x):=4*x[1]+x[2]-x[3]+2*x[4]-x[5];

(%o3) $\mathrm{F}(x):=4\,x_1+x_2-x_3+2\,x_4-x_5$

(%i4) sist:[eq1,eq2]$

(%i5) vars:listofvars(sist);

(%o5) $[x_1, x_2, x_3, x_4, x_5]$

(%i6) Ai:augcoefmatrix(sist,vars);

(%o6) $\begin{bmatrix} 3 & 1 & 2 & -1 & 0 & -5 \\ 2 & 0 & -1 & 3 & 4 & -8 \end{bmatrix}$

```
(%i7) A:submatrix(Ai,6);
```

(%o7) $\begin{bmatrix} 3 & 1 & 2 & -1 & 0 \\ 2 & 0 & -1 & 3 & 4 \end{bmatrix}$

```
(%i8) [m,n]:matrix_size(A);
```

(%o8) $[2,5]$

Matricos A rangas:

```
(%i9) r:rank(A);
```

(%o9) 2

Laivųjų nežinomųjų skaičius:

```
(%i10) k:n-r;
```

(%o10) 3

```
(%i11) var:full_listify(powerset(setify(vars), k));
```

(%o11) $[[x_1,x_2,x_3],[x_1,x_2,x_4],[x_1,x_2,x_5],[x_1,x_3,x_4],[x_1,x_3,x_5],[x_1,x_4,x_5],[x_2,x_3,x_4],[x_2,x_3,x_5],[x_2,x_4,x_5],[x_3,x_4,x_5]]$

```
(%i12) makelist(solve(append(sist,var[i]),vars),i,1,length(var));
```

(%o12) $[[[x_1=0,x_2=0,x_3=0,x_4=-5,x_5=\frac{23}{4}]],[[x_1=0,x_2=0,x_3=\frac{5}{2},x_4=0,x_5=\frac{21}{8}]],[[x_1=0,x_2=0,x_3=\frac{23}{5},x_4=\frac{21}{5},x_5=0]],[[x_1=0,x_2=5,x_3=0,x_4=0,x_5=2]],[[x_1=0,x_2=\frac{23}{3},x_3=0,x_4=\frac{8}{3},x_5=0]],[[x_1=0,x_2=21,x_3=-8,x_4=0,x_5=0]],[[x_1=\frac{5}{3},x_2=0,x_3=0,x_4=0,x_5=\frac{7}{6}]],[[x_1=\frac{23}{11},x_2=0,x_3=0,x_4=\frac{14}{11},x_5=0]],[[x_1=3,x_2=0,x_3=-2,x_4=0,x_5=0]],[[x_1=4,x_2=-7,x_3=0,x_4=0,x_5=0]]]$

Radome visus bazinius sprendinius:

```
(%i13) baz:makelist(subst(%[i],vars),i,1,length(%));
```

(%o13) $[[0,0,0,-5,\frac{23}{4}],[0,0,\frac{5}{2},0,\frac{21}{8}],[0,0,\frac{23}{5},\frac{21}{5},0],[0,5,0,0,2],[0,\frac{23}{3},0,\frac{8}{3},0],[0,21,-8,0,0],[\frac{5}{3},0,0,0,\frac{7}{6}],[\frac{23}{11},0,0,\frac{14}{11},0],[3,0,-2,0,0],[4,-7,0,0,0]]$

Leistinosios aibės X kraštutiniai taškai yra:

```
(%i14) krast:sublist(baz,lambda([e],lmin(e)>=0));
```

(%o14) $[[0,0,\frac{5}{2},0,\frac{21}{8}],[0,0,\frac{23}{5},\frac{21}{5},0],[0,5,0,0,2],[0,\frac{23}{3},0,\frac{8}{3},0],[\frac{5}{3},0,0,0,\frac{7}{6}],[\frac{23}{11},0,0,\frac{14}{11},0]]$

Randame tikslo funkcijos reikšmes kraštiniuose taškuose:

```
(%i15) map(F,krast);
```

(%o15) $[-\frac{41}{8},\frac{19}{5},3,13,\frac{11}{2},\frac{120}{11}]$

Mažiausioji reikšmė yra:

```
(%i16) min_r:lmin(%);
```

(%o16) $-\frac{41}{8}$

Ji įgyjama taškuose(šiuo atveju yra tik vienas taškas):

```
(%i17) sublist(krast,lambda([x],F(x)=min_r));
```

(%o17) $[[0,0,\frac{5}{2},0,\frac{21}{8}]]$

4 būdas. Pasinaudosime programa "ext"((C), A.Domarkas) kraštutinių taškų radimui:

```
(%i18) kill(all)$
```

```
(%i1) load(simplex)$
```

```
(%i2) ext(apr):=block([var,fs,cs,ap,s,S,m],
      var:sort(listofvars(apr)),
      s:apply("+",var),
      fs:append([1,s,-s],var,-var),
      ap(k):=subst(apr[k]=(lhs(apr[k])=rhs(apr[k])),apr),
      cs:makelist(ap(k),k,1,length(apr)),
      S:[],
      for f in fs do
      for c in cs do
      (
      m:minimize_lp(f,c),
      if listp(m) then
      S:cons(subst(m[2],var),S)
      ),
      listify(setify(S))
      )$
```

```
(%i3) F(x):=4*x[1]+x[2]-x[3]+2*x[4]-x[5];
```

(%o3) $\mathrm{F}(x):=4\,x_1+x_2-x_3+2\,x_4-x_5$

```
(%i4) apr:[3*x[1]+x[2]+2*x[3]-x[4]=5, 2*x[1]-x[3]+3*x[4]+4*x[5]=8,
      x[1]>=0,x[2]>=0,x[3]>=0,x[4]>=0,x[5]>=0]$
```

Randame kraštutinius taškus:

```
(%i5) krast:ext(apr);
```

(%o5) $[[0,0,\frac{5}{2},0,\frac{21}{8}],[0,0,\frac{23}{5},\frac{21}{5},0],[0,5,0,0,2],[0,\frac{23}{3},0,\frac{8}{3},0],[\frac{5}{3},0,0,0,\frac{7}{6}],[\frac{23}{11},0,0,\frac{14}{11},0]]$

Randame tikslo funkcijos reikšmes kraštutiniuose taškuose:

```
(%i6) map(F,krast);
```

(%o6) $[-\frac{41}{8},\frac{19}{5},3,13,\frac{11}{2},\frac{120}{11}]$

Mažiausioji reikšmė yra:

```
(%i7) min_r:lmin(%);
```

(%o7) $-\frac{41}{8}$

Ji įgyjama taškuose(šiuo atveju yra tik vienas taškas):

```
(%i8) sublist(krast,lambda([x],F(x)=min_r));
```

(%o8) $[[0,0,\frac{5}{2},0,\frac{21}{8}]]$

5 būdas. Pasinaudosime programa "nopt"((C), A.Domarkas) netiesinės optimizacijos uždavinių sprendimui. Ši programa sprendžia ir tiesinio programavimo uždavinius su nedideliu kintamųjų skaičiumi. Joje nenaudojamas simplekso metodas, bet naudojamas Lagranžo daugiklių metodas.

```
(%i9) kill(all)$ reset();
```
(%o1) [*features*, *piece*, _, __, *labels*, %, *tr-unique*, *lispdisp*, *multiplicities*, *pivot_count_sx*]

```
(%i2) load(nopt);
```
(%o2) *C:/Users/Aleksas/maxima/nopt.mac*

```
(%i3) f:4*x1+x2-x3+2*x4-x5$
```

```
(%i4) apr:[3*x1+x2+2*x3-x4=5, 2*x1-x3+3*x4+4*x5=8,
      x1>=0,x2>=0,x3>=0,x4>=0,x5>=0]$
```

```
(%i5) minimize_nopt(f,apr);
```
(%o5) $\left[-\frac{41}{8},\left[x1=0,x2=0,x3=\frac{5}{2},x4=0,x5=\frac{21}{8}\right]\right]$

6 būdas. Žr. [1], 294 p.

Palyginkite sprendimo būdus.
Ar racionalu šiais laikais tokius uždavinius spręsti ranka be kompiuterio?

5 pavyzdys

5 pavyzdys([5], Example 6.24).
Neneigiamų sprendinių aibėje rasti funkcijos f minimummą, kai
teisingi apribojimai g.

Figure 1:

$$f = 9/2x_{81} + 1/2x_{82} + x_{83} + 17/2x_{121} + 5x_{122} + 3/2x_{123} + 9/2x_{124} + x_{125};$$
$$g = \{x_{81} + 2x_{83} + x_{121} + 2x_{122} + 3x_{123} + x_{125} \ge 25, x_{82} + x_{124} + x_{125} \ge 35\};$$

Šiame pavyzdyje uždavinys turi kelis sprendinius,
be to, nenaudojamas suvedimas į kanoninį pavidalą.

1 būdas.

```
(%i6) load(simplex)$
```

```
(%i7) f:9*x1/2+x2/2+x3+17*x4/2+5*x5+3*x6/2+9*x7/2+x8;
```
(%o7) $x8+\frac{9\,x7}{2}+\frac{3\,x6}{2}+5\,x5+\frac{17\,x4}{2}+x3+\frac{x2}{2}+\frac{9\,x1}{2}$

```
(%i8) apr:[x1+2*x3+x4+2*x5+3*x6+x8>=25,x2+x7+x8>=35];
```
(%o8) $[x8+3\,x6+2\,x5+x4+2\,x3+x1>=25, x8+x7+x2>=35]$

```
(%i9) minimize_lp(f,apr),nonegative_lp=true;
```
(%o9) $\left[30,\left[x7=0,x2=35,x8=0,x6=\frac{25}{3},x5=0,x4=0,x3=0,x1=0\right]\right]$

```
(%i10) sort(%[2]);
```
(%o10) $\left[x1=0,x2=35,x3=0,x4=0,x5=0,x6=\frac{25}{3},x7=0,x8=0\right]$

2 būdas. Pasinaudosime programa "nopt"((C), A.Domarkas) netiesinės
optimizacijos uždavinių sprendimui. Ši programa sprendžia ir tiesinio
programavimo uždavinius su nedideliu kintamųjų skaičiumi.

```
(%i11) load(nopt);
```
(%o11) *C:/Users/Aleksas/maxima/nopt.mac*

```
(%i12) vars:listofvars([f,apr]);
```
(%o12) $[x1,x2,x3,x4,x5,x6,x7,x8]$

```
(%i13) nn:makelist(vars[i]>=0,i,length(vars));
```

(%o13) $[x1>=0, x2>=0, x3>=0, x4>=0, x5>=0, x6>=0, x7>=0, x8>=0]$

```
(%i14) apr1:append(apr,nn);
```

(%o14) $[x8+3\,x6+2\,x5+x4+2\,x3+x1>=25, x8+x7+x2>=35, x1>=0, x2>=0, x3>=0, x4>=0, x5>=0, x6>=0, x7>=0, x8>=0]$

```
(%i15) spr:minimize_nopt(f,apr1);
```

(%o15) $[30, [x1=0, x2=35, x3=\frac{25}{2}, x4=0, x5=0, x6=0, x7=0, x8=0], [x1=0, x2=35, x3=0, x4=0, x5=0, x6=\frac{25}{3}, x7=0, x8=0], [x1=0, x2=10, x3=0, x4=0, x5=0, x6=0, x7=0, x8=25]]$

Gavome tris sprendinius.
Visi jie tinka, nes juos įstačius į f gauname tą pačią reikšmę:

```
(%i16) subst(spr[2],f);
```

(%o16) 30

```
(%i17) subst(spr[3],f);
```

(%o17) 30

```
(%i18) subst(spr[4],f);
```

(%o18) 30

3 būdas. Pasinaudosime programa "ext"((C), A.Domarkas) kraštutinių taškų radimui:

```
(%i19) ext(apr):=block([var,fs,cs,ap,s,S,m],
       var:sort(listofvars(apr)),
       s:apply("+",var),
       fs:append([1,s,-s],var,-var),
       ap(k):=subst(apr[k]=(lhs(apr[k])=rhs(apr[k])),apr),
       cs:makelist(ap(k),k,1,length(apr)),
       S:[],
       for f in fs do
       for c in cs do
       (
       m:minimize_lp(f,c),
       if listp(m) then
       S:cons(subst(m[2],var),S)
       ),
       listify(setify(S))
       )$
```

```
(%i20) krast:ext(apr1);
```

(%o20) $[[0,0,0,0,0,0,0,35], [0,0,0,0,0,0,10,25], [0,0,0,0,0,\frac{25}{3},35,0], [0,0,\frac{25}{2},0,0,0,35,0], [0,10,0,0,0,0,0,25], [0,35,0,0,0,\frac{25}{3},0,0], [0,35,0,0,\frac{25}{2},0,0,0], [0,35,0,25,0,0,0,0], [0,35,\frac{25}{2},0,0,0,0,0], [25,35,0,0,0,0,0,0]]$

Apskaičiuosime tikslo funkcijos reikšmes šiuose taškuose.
Tam šiek tiek pakeisime tikslo funkcijos įvedimą.

```
(%i21) define(g(x),subst(makelist(concat(x,i)=x[i],i,1,8),f));
```

(%o21) $g(x):=x_8+\frac{9\,x_7}{2}+\frac{3\,x_6}{2}+5\,x_5+\frac{17\,x_4}{2}+x_3+\frac{x_2}{2}+\frac{9\,x_1}{2}$

Randame tikslo funkcijos reikšmes kraštutiniuose taškuose:

```
(%i22) map(g,krast);
```

(%o22) $[35, 70, 170, 170, 30, 30, 80, 230, 30, 130]$

Mažiausioji reikšmė yra:

```
(%i23) min_r:lmin(%);
(%o23) 30
```

Ji įgyjama taškuose:

```
(%i24) sublist(krast,lambda([x],g(x)=min_r));
```

(%o24) $[[0,10,0,0,0,0,0,25],[0,35,0,0,0,\frac{25}{3},0,0],[0,35,\frac{25}{2},0,0,0,0,0]]$

Mathematica(žr, [5], Example 6.24) ir Maxima simplex paketas(žr. 1 sprendimo būdą) gauna tik vieną sprendinį.

6 Literatūra

[1] A. Apynis, E. Stankus, Matematikos pagrindai, V., TEV, 2009
[2] A. Apynis, E. Stankus, Matematika, V., TEV, 2001
[3] A. Apynis, Optimizavimo metodai, V., VU, 2005
[4] V. Čiočys, R.Jasiulionis, Matematinis programavimas, V., Mokslas, 1990
[5] M. Asghar Bhatti, Practical optimization methods: with Mathematica Aplications, Springer, 2