Vieno kintamojo funkcijos ekstremumai

- Funkcijos y = f(x) apibrėžimo srities vidinis taškas x = c yra vadinamas funkcijos kritiniu tašku, jei f'(c) = 0 arba f'(c) neegzistuoja.
- Jei f'(c) = 0 ir f''(c) > 0, tai c - lokalaus minimumo taškas.
- Jei f'(c) = 0 ir f''(c) < 0, tai c - lokalaus maksimumo taškas.
- Jei f'(c) = 0 ir f''(c) = 0, tai reikalingas papildomas tyrimas.

Čia naudojama atvirojo kodo kompiuterinės algebros programa Maxima-sbcl-5.36.1

Kritinių taškų analizei pagal antrosios išvestinės ženklą apibrėžiame komandą test:

```
(%i1) test(sol):=block(minp:[], maxp:[],
      for k thru length(sol) do
      (
      if freeof(%i,sol[k]) and at(diff(f(x),x,2),sol[k])>0 then
      (print ("minimumas: ", ev('f(x)=f(x),sol[k])),minp:endcons(ev([x,f(x)],sol[k]),minp))
      elseif freeof(%i,sol[k]) and at(diff(f(x),x,2),sol[k])<0 then
      (print ("maksimumas: ", ev('f(x)=f(x),sol[k])),maxp:endcons(ev([x,f(x)],sol[k]),maxp))
      ))$
```

1 pavyzdys. Rasti funkcijos y = x^4 - x   lokaliuosius ekstremumus.

```
(%i2) f(x):=x^4-x$
```

```
(%i3) eq:diff(f(x),x)=0;
```

(%o3) $4\,x^3-1=0$

```
(%i4) sol:solve(eq);
```

(%o4) $[x=\frac{\sqrt{3}\,\%i-1}{2\,4^{1/3}},x=-\frac{\sqrt{3}\,\%i+1}{2\,4^{1/3}},x=\frac{1}{4^{1/3}}]$

```
(%i5) test(sol)$
```

minimumas: $f\left(\frac{1}{4^{1/3}}\right)=-\frac{3}{4^{4/3}}$

Arba skaitiškai:

```
(%i6) test(sol),numer$
```

minimumas: $f(0.6299605249474366)=-0.4724703937105774$

```
(%i7) minp;
```

(%o7) $[[0.6299605249474366,-0.4724703937105774]]$

```
(%i8) wxplot2d([f(x),['discrete, minp]],
      [x,-0.1,1.2],
      [style,[lines,1,7], [points,2,2,1]],
      [legend, "y = x^4 - x", "minimumas"]
      )$
```

(%t8)



2 pavyzdys

```
(%i9) f(x):=x^4/4-2*x^2$
```

```
(%i10) sol:solve(diff(f(x),x)=0,x);
```

(%o10) $[x=-2, x=2, x=0]$

```
(%i11) test(sol)$
```

minimumas: $f(-2)=-4$

minimumas: $f(2)=-4$

maksimumas: $f(0)=0$

```
(%i12) minp;
```

(%o12) $[[-2,-4],[2,-4]]$

```
(%i13) maxp;
```

(%o13) $[[0,0]]$

(%i14)
```
wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,-3,3],
[style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
[legend, string(y = f(x)), "minimumas", "maksimumas"]
)$
```

(%t14)



3 pavyzdys. Rasti funkcijos y = 10*x^6+12*x^5+15*x^4+20*x^3-360*x^2-720*x lokaliuosius ekstremumus.

(%i15)
```
f(x):=10*x^6+12*x^5+15*x^4+20*x^3-360*x^2-720*x$
```

(%i16)
```
eq:diff(f(x),x)=0;
```

(%o16) $60\,x^5+60\,x^4+60\,x^3+60\,x^2-720\,x-720=0$

(%i17)
```
sol:solve(eq,x);
```

(%o17) $[x=-\sqrt{3}, x=\sqrt{3}, x=-1, x=-2\,\%i, x=2\,\%i]$

Jei lygties eq kairioji pusė yra daugianaris ir komanda solve tos lygties neišsprendžia, tai vietoj solve naudokite allroots:

(%i18)
```
allroots(eq);
```

(%o18) $[x=1.732050807568877, x=-0.9999999999999999, x=-1.732050807568877, x=2.0\,\%i, x=-2.0\,\%i]$

(%i19)
```
test(sol)$
```

minimumas: $f\left(-\sqrt{3}\right)=184\,3^{3/2}-675$

minimumas: $f\left(\sqrt{3}\right)=-184\,3^{3/2}-675$

maksimumas: $f\left(-1\right)=353$

(%i20)
```
minp;
```

(%o20) $[[-\sqrt{3}, 184\,3^{3/2}-675], [\sqrt{3}, -184\,3^{3/2}-675]]$

(%i21)
```
maxp;
```

(%o21) $[[-1, 353]]$

Arba skaitiškai:

(%i22)
```
test(sol),numer$
```

minimumas: $f\left(-1.732050807568877\right)=281.0920457780204$

minimumas: $f\left(1.732050807568877\right)=-1631.09204577802$

maksimumas: $f\left(-1\right)=353$

```
(%i23) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,-3,3],[y,-2000,2000],
       [style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
       [legend, "y = f(x)", "minimumas", "maksimumas"],
       [gnuplot_preamble,"set key top center;"]
       )$
plot2d: some values were clipped.
```

(%t23)



4 pavyzdys Rasti funkcijos y = x*(1-x)*exp(x) lokaliuosius ekstremumus.

```
(%i24) f(x):=x*(1-x)*exp(x)$
(%i25) sol:solve(diff(f(x),x)=0,x);
```

(%o25) $[x=-\frac{\sqrt{5}+1}{2}, x=\frac{\sqrt{5}-1}{2}]$

```
(%i26) test(sol)$
```

minimumas: $f\left(-\frac{\sqrt{5}+1}{2}\right)=-\frac{\left(\sqrt{5}+1\right)\left(\frac{\sqrt{5}+1}{2}+1\right)\%e^{-\frac{\sqrt{5}+1}{2}}}{2}$

maksimumas: $f\left(\frac{\sqrt{5}-1}{2}\right)=\frac{\left(1-\frac{\sqrt{5}-1}{2}\right)\left(\sqrt{5}-1\right)\%e^{\frac{\sqrt{5}-1}{2}}}{2}$

Minimumo taškas:

```
(%i27) minp,expand,factor;
```

(%o27) $[[-\frac{\sqrt{5}+1}{2}, -\left(\sqrt{5}+2\right)\%e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}}]]$

Maksimumo taškas:

```
(%i28) maxp,expand,factor;
```

(%o28) $[[\frac{\sqrt{5}-1}{2}, \left(\sqrt{5}-2\right)\%e^{\frac{\sqrt{5}}{2}-\frac{1}{2}}]]$

Arba skaitiškai:

```
(%i29) test(sol),numer$
```

minimumas: $f(-1.618033988749895)=-0.8399620946571751$

maksimumas: $f(0.6180339887498949)=0.43797147932204$

```
(%i30) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,-3,1.2],
       [style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
       [legend, string(y = f(x)), "minimumas", "maksimumas"],
       [gnuplot_preamble,"set key top left;"]
       )$
```

(%t30)



5 pavyzdys Rasti funkcijos y = sin(x)*(1+cos(x)) lokaliuosius ekstremumus.

```
(%i31) f(x):=sin(x)*(1+cos(x))$
```

```
(%i32) diff(f(x),x)=0;
```

(%o32) $\cos(x)\left(\cos(x)+1\right)-\sin(x)^2=0$

```
(%i33) eq:trigrat(%);
```

(%o33) $\cos(2\,x)+\cos(x)=0$

```
(%i34) load(trigtools);
```

(%o34)
C:/Program Files (x86)/Maxima-sbcl-5.36.1/share/maxima/5.36.1/share/contrib/trigtools/trigtools.m

paketo "trigtools " autorius - A.Domarkas

```
(%i35) trigsolve(eq,0,2*%pi);
```

to_poly_solve: to_poly_solver.mac is obsolete; I'm loading to_poly_solve.mac instead.

(%o35) $\{\frac{\pi}{3}, \pi, \frac{5\,\pi}{3}\}$

```
(%i36) listify(%);
```

(%o36) $[\frac{\pi}{3}, \pi, \frac{5\,\pi}{3}]$

```
(%i37) sol:makelist(x=%[k],k,1,length(%));
```

(%o37) $[x=\frac{\pi}{3}, x=\pi, x=\frac{5\,\pi}{3}]$

```
(%i38) test(sol);
```

maksimumas: $f\left(\frac{\pi}{3}\right)=\frac{3^{3/2}}{4}$

minimumas: $f\left(\frac{5\,\pi}{3}\right)=-\frac{3^{3/2}}{4}$

(%o38) *done*

(%i39) minp;

(%o39) $[[\frac{5\pi}{3}, -\frac{3^{3/2}}{4}]]$

(%i40) maxp;

(%o40) $[[\frac{\pi}{3}, \frac{3^{3/2}}{4}]]$

```
(%i41) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,0,2*%pi],
       [style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
       [legend, string(y = f(x)), "minimumas", "maksimumas"]
       )$
```

(%t41)



6 pavyzdys. Rasime pirmosios rūšies 5-os eilės Čebyšovo polinomo
T[5] = 16*x^5 - 20*x^3 + 5*x lokaliuosius ekstremumus:

(%i42) load(orthopoly)$

Reikšmių suprastinimui panaudosime paketą "sqdnst" ir jo komandą "sqrtdenest".

(%i43) load (sqdnst)$

(%i44) f(x):=expand(chebyshev_t (5, x));

(%o44) $\mathrm{f}(x):=\mathrm{expand}\left(T_5(x)\right)$

(%i45) f(x);

(%o45) $16x^5-20x^3+5x$

(%i46) solve(diff(f(x),x)=0,x);

(%o46) $[x=-\frac{\sqrt{\sqrt{5}+3}}{2^{3/2}}, x=\frac{\sqrt{\sqrt{5}+3}}{2^{3/2}}, x=-\frac{\sqrt{3-\sqrt{5}}}{2^{3/2}}, x=\frac{\sqrt{3-\sqrt{5}}}{2^{3/2}}]$

(%i47) sol:ratsimp(sqrtdenest(%));

(%o47) $[x=-\frac{\sqrt{5}+1}{4}, x=\frac{\sqrt{5}+1}{4}, x=-\frac{\sqrt{5}-1}{4}, x=\frac{\sqrt{5}-1}{4}]$

(%i48) `test(sol);`

maksimumas: $f\left(-\frac{\sqrt{5}+1}{4}\right)=1$

minimumas: $f\left(\frac{\sqrt{5}+1}{4}\right)=-1$

minimumas: $f\left(-\frac{\sqrt{5}-1}{4}\right)=-1$

maksimumas: $f\left(\frac{\sqrt{5}-1}{4}\right)=1$

(%o48) *done*

Minimumo taškai:

(%i49) `minp;`

(%o49) $\left[\left[\frac{\sqrt{5}+1}{4},-1\right],\left[-\frac{\sqrt{5}-1}{4},-1\right]\right]$

Maksimumo taškai:

(%i50) `maxp;`

(%o50) $\left[\left[-\frac{\sqrt{5}+1}{4},1\right],\left[\frac{\sqrt{5}-1}{4},1\right]\right]$

(%i51)
```
wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,-1,1],[y,-1,2],
[style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
[legend, string(y = f(x)), "minimumas", "maksimumas"]
)$
```

(%t51)



Ištirkite kitus Čebyšovo polinomus.

7 pavyzdys. Rasti funkcijos y = x^5-3x^4+5 maksimumą ir minimumą,
kai x priklauso intervalui [0, 4].

(%i52) `f(x):=x^5-3*x^4+5$`

```
(%i53) wxplot2d([f(x)], [x,0,4])$
```

(%t53)



Iš brėžinio matome, kad globalus maksimumas, kai x priklauso [0, 4], yra pasiekiamas taške x = 4.

```
(%i54) f(4);
```

(%o54) 261

Todėl f_max =f(4) = 261

```
(%i55) eq:diff(f(x),x)=0;
```

(%o55) $5\,x^4-12\,x^3=0$

```
(%i56) sol:solve(eq);
```

(%o56) $[x=\frac{12}{5},x=0]$

```
(%i57) test(sol)$
```

minimumas: $\mathrm{f}\left(\frac{12}{5}\right)=-\frac{46583}{3125}$

```
(%i58) test(sol),numer$
```

minimumas: $\mathrm{f}(2.4)=-14.90656000000001$

Minimumo taškas:

```
(%i59) minp;
```

(%o59) [[2.4,-14.90656000000001]]

Globalaus maksimumo taškas:

```
(%i60) maxp:[[4, 261]];
```

(%o60) [[4,261]]

```
(%i61) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,0,4],[y,-20,300],
       [style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
       [legend, string(y = f(x)), "minimumas", "maksimumas"]
       )$
```

(%t61)



Sprendimas su paketu "nopt":

```
(%i62) load(nopt);
```

(%o62) *C:/Users/aleksas/maxima/nopt.mac*

```
(%i63) maximize_nopt(f(x),[x>=0,x<=4]);
```

(%o63) $[261,[x=4]]$

```
(%i64) minimize_nopt(f(x),[x>=0,x<=4]);
```

(%o64) $[-\frac{46583}{3125},[x=\frac{12}{5}]]$

8 pavyzdys. Rasti funkcijos y = x*(1-x)*exp(x) lokaliuosius ekstremumus intervale [0, 20].

```
(%i65) f(x):=sqrt(x)*cos(x)*erf(x);
```

(%o65) $\mathrm{f}(x):=\sqrt{x}\cos(x)\,\mathrm{erf}(x)$

```
(%i66) f1:diff(f(x),x);
```

(%o66) $-\sqrt{x}\,\mathrm{erf}(x)\sin(x)+\frac{\cos(x)\,\mathrm{erf}(x)}{2\sqrt{x}}+\frac{2\sqrt{x}\,\%e^{-x^2}\cos(x)}{\sqrt{\pi}}$

Su solve ar allroots lygties f'(x) = 0 išspręsti nepavyks.
Todėl skaitiniam sprendimui naudojame find_root.

```
(%i67) wxplot2d([f], [x,0,20])$
```

(%t67)



Iš brėžinio matome, kad intervale [0, 20] funkcija turi 7 lokaliuosius ekstremumus, kurie yra intervaluose [0.1,2], [2,5], [5, 8], [8,11], [11,14], [14, 17], [17, 20].
Todėl dalijimo taškai yra:

```
(%i68) X:[0.1, 2, 5, 8, 11, 14, 17, 20];
(%o68) [0.1,2,5,8,11,14,17,20]
```

Kritiniai taškai:

```
(%i69) kt:makelist(x=find_root(f1, x, X[k], X[k+1]),k,1,7);
(%o69) [x=0.8841855120621396,x=3.292330717436803,x=6.361620392065665,x=
9.477485705420795,x=12.60601344427541,x=15.73971935600487,x=18.87603833798585]
```

```
(%i70) test(kt);
maksimumas: f(0.8841855120621396)=0.4702230129764967
minimumas: f(3.292330717436803)=-1.793897052645326
maksimumas: f(6.361620392065665)=2.514470818617907
minimumas: f(9.477485705420795)=-3.074277250870972
maksimumas: f(12.60601344427541)=3.547705285073695
minimumas: f(15.73971935600487)=-3.965331257867859
maksimumas: f(18.87603833798585)=4.343132892252145
(%o70) done
```

```
(%i71) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,0,20],[y,-5,7],
       [style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
       [legend, string(y = f(x)), "minimumas", "maksimumas"]
       )$
```

(%t71)



9 pavyzdys. Rasti funkcijos
y = 10*x^7-35*x^6-42*x^5+210*x^4-70*x^3-210*x^2+140*x+2
lokaliuosius ekstremumus

```
(%i72) f(x):=10*x^7-35*x^6-42*x^5+210*x^4-70*x^3-210*x^2+140*x+2$
```

```
(%i73) f1:diff(f(x),x);
```

(%o73) $70\,x^6-210\,x^5-210\,x^4+840\,x^3-210\,x^2-420\,x+140$

Lygtį f'(x) = 0 tiksliai pavyksta išspręsti su komanda solvet iš paketo odes:

```
(%i74) load(odes);
```

(%o74)
C:/Program Files (x86)/Maxima-sbcl-5.36.1/share/maxima/5.36.1/share/contrib/odes/odes.mac

```
(%i75) kt:solvet(f1,x);
```

(%o75) $[x=1,x=1-\sqrt{3},x=\sqrt{3}+1,x=2\cos\left(\frac{2\pi}{9}\right),x=2\cos\left(\frac{4\pi}{9}\right),x=2\cos\left(\frac{8\pi}{9}\right)]$

Skaitiškai visas šaknis galima rasti su komanda allroots:

```
(%i76) allroots(f1);
```

(%o76) $[x=0.3472963553338607,x=-0.7320508075688773,x=1.0,x=1.532088886237956,x=-1.879385241571817,x=2.732050807568876]$

(%i77) test(kt)$

minimumas: $f(1)=5$

minimumas: $f(1-\sqrt{3})=140\,(1-\sqrt{3})+10\,(1-\sqrt{3})^7-35\,(1-\sqrt{3})^6-42\,(1-\sqrt{3})^5+210\,(1-\sqrt{3})^4-70\,(1-\sqrt{3})^3-210\,(1-\sqrt{3})^2+2$

minimumas: $f(\sqrt{3}+1)=10\,(\sqrt{3}+1)^7-35\,(\sqrt{3}+1)^6-42\,(\sqrt{3}+1)^5+210\,(\sqrt{3}+1)^4-70\,(\sqrt{3}+1)^3-210\,(\sqrt{3}+1)^2+140\,(\sqrt{3}+1)+2$

maksimumas: $f\left(2\cos\left(\frac{2\pi}{9}\right)\right)=1280\cos\left(\frac{2\pi}{9}\right)^7-2240\cos\left(\frac{2\pi}{9}\right)^6-1344\cos\left(\frac{2\pi}{9}\right)^5+3360\cos\left(\frac{2\pi}{9}\right)^4-560\cos\left(\frac{2\pi}{9}\right)^3-840\cos\left(\frac{2\pi}{9}\right)^2+280\cos\left(\frac{2\pi}{9}\right)+2$

maksimumas: $f\left(2\cos\left(\frac{4\pi}{9}\right)\right)=1280\cos\left(\frac{4\pi}{9}\right)^7-2240\cos\left(\frac{4\pi}{9}\right)^6-1344\cos\left(\frac{4\pi}{9}\right)^5+3360\cos\left(\frac{4\pi}{9}\right)^4-560\cos\left(\frac{4\pi}{9}\right)^3-840\cos\left(\frac{4\pi}{9}\right)^2+280\cos\left(\frac{4\pi}{9}\right)+2$

maksimumas: $f\left(2\cos\left(\frac{8\pi}{9}\right)\right)=1280\cos\left(\frac{8\pi}{9}\right)^7-2240\cos\left(\frac{8\pi}{9}\right)^6-1344\cos\left(\frac{8\pi}{9}\right)^5+3360\cos\left(\frac{8\pi}{9}\right)^4-560\cos\left(\frac{8\pi}{9}\right)^3-840\cos\left(\frac{8\pi}{9}\right)^2+280\cos\left(\frac{8\pi}{9}\right)+2$

Funkcijos simbolines reikšmes galima suprastinti. Atlikite tai patys.
Skaitinės reikšmės yra:

(%i78) test(kt),numer$

minimumas: $f(1)=5$

minimumas: $f(-0.7320508075688772)=-122.9385127825612$

minimumas: $f(2.732050807568877)=-497.0614872174392$

maksimumas: $f(1.532088886237956)=19.82385002606929$

maksimumas: $f(0.3472963553338608)=25.14769628716334$

maksimumas: $f(-1.879385241571817)=696.0284536867675$

(%i79) wxplot2d([f(x),['discrete, minp],['discrete, maxp]], [x,-2.3, 3.2],[y,-500,850],
[style,[lines,1,7], [points,3,2,1],[points,3,3,1]],
[legend, string(y = f(x)), "minimumas", "maksimumas"]
)$

plot2d: some values were clipped.

(%t79)