TP taikymo uždaviniai

A.Domarkas, VU

Čia naudojama atvirojo kodo kompiuterinės algebros sistema Maxima 5.31.2

1 *Transporto uždavinys*

1 pavyzdys.([3], 374 p.) Firma, turinti 3 degalines, benziną perka dviejose naftos bazėse.
Reikia sudaryti pervežimo planą, kurio pervežimo kaštai būtų mažiausi,
jei yra nurodytos pervežimo kainos(matrica K), benzino poreikis(sarašas po)
ir atsargos(sąrašas at).

Pervežimo kainos:

```
(%i1) K:matrix([20,30,40],[36,22,28]);
```

$$(\%o1) \quad \begin{bmatrix} 20 & 30 & 40 \\ 36 & 22 & 28 \end{bmatrix}$$

Poreikiai:

```
(%i2) po:[6,9,7]; n:length(po);
(%o2) [6,9,7]
(%o3) 3
```

Atsargos:

```
(%i4) at:[10,12]; m:length(at);
(%o4) [10,12]
(%o5) 2
```

Pervežimų matrica:

```
(%i6) X: genmatrix(lambda([i,j], x[concat(i,j)]), 2, 3);
```

$$(\%o6) \quad \begin{bmatrix} x_{11} & x_{12} & x_{13} \\ x_{21} & x_{22} & x_{23} \end{bmatrix}$$

x[i,j] yra uždavinio nežinomieji(rodo kiek reikia pervezti iš i-osios bazės į j-ają degalinę).

Visos benzino atsargos iš naftos bazių turi būti išvežtos. Todėl:

```
(%i7) L1:makelist(sum(X[i,j],j,1,n)=at[i],i,1,m);
```

$$(\%o7) \quad [x_{13}+x_{12}+x_{11}=10, x_{23}+x_{22}+x_{21}=12]$$

Visų degalinių poreikiai turi būti patenkinti. Todėl:

```
(%i8) L2:makelist(sum(X[i,j],i,1,m)=po[j],j,1,n);
```
(%o8) $[x_{21}+x_{11}=6, x_{22}+x_{12}=9, x_{23}+x_{13}=7]$

```
(%i9) apr:join(L1,L2);
```
(%o9) $[x_{13}+x_{12}+x_{11}=10, x_{21}+x_{11}=6, x_{23}+x_{22}+x_{21}=12, x_{22}+x_{12}=9]$

Benzino pervežimo kaštai yra nusakomi tikslo funkcija

```
(%i10) f:sum(sum(K[i,j]*X[i,j],i,1,m),j,1,n);
```
(%o10) $28\, x_{23}+22\, x_{22}+36\, x_{21}+40\, x_{13}+30\, x_{12}+20\, x_{11}$

Sprendžiame simplekso metodu:

```
(%i11) load(simplex)$
```

```
(%i12) spr:minimize_lp(f,apr),nonegative_lp=true;
```
(%o12) $[546, [x_{23}=7, x_{22}=5, x_{21}=0, x_{13}=0, x_{12}=4, x_{11}=6]]$

```
(%i13) reverse(spr[2]);
```
(%o13) $[x_{11}=6, x_{12}=4, x_{13}=0, x_{21}=0, x_{22}=5, x_{23}=7]$

Optimalioji pervežimų matrica yra

```
(%i14) X=subst(spr[2],X);
```
(%o14) $$\begin{bmatrix} x_{11} & x_{12} & x_{13} \\ x_{21} & x_{22} & x_{23} \end{bmatrix} = \begin{bmatrix} 6 & 4 & 0 \\ 0 & 5 & 7 \end{bmatrix}$$

Ats. f_min=546, kai x_11=6, x_12=4, x_13=0, x_21=0, x_22=5, x_23=7 .

2 Gamybos uždavinys

Pavyzdys([2], 67 p.). Stalius ruošiasi daryti kėdes ir taburetes. Be visų kitų medžiagų, jam reikalingos keturių rūšių lentų išpjovos, kurių atsargos išreikštos sąraše at = [20, 14, 16, 18].
Už kėdę panuojama gauti 30 Lt, o už taburetę - 20 Lt. Išpjovų sunaudojimo lentelė yra išreiškiama matrica

```
(%i15) A:matrix([3,2],[2,0],[1,1],[1,2]);
```
(%o15) $$\begin{bmatrix} 3 & 2 \\ 2 & 0 \\ 1 & 1 \\ 1 & 2 \end{bmatrix}$$

Čia matricos A pirmasis stulpelis nurodo kiek išpjovų sunaudojama vienai kėdei.
Matricos A antrasis stulpelis nurodo kiek išpjovų sunaudojama vienai taburetei.
Kiek kėdžių ir taburečių reiktų padaryti staliui, norint gauti didžiausią pelną?

Sprendimas.

```
(%i16) at:transpose([20, 14, 16, 18]);
```

(%o16) $\begin{bmatrix} 20 \\ 14 \\ 16 \\ 18 \end{bmatrix}$

Tarkime, kad x1 - kėdžių skaičius, x2 - taburečių.
Tada pajamos išreškiamos tikslo funkcija

```
(%i17) f:30*x1 + 20*x2;
```

(%o17) $20\,x2+30\,x1$

```
(%i18) X:transpose([x1,x2]);
```

(%o18) $\begin{bmatrix} x1 \\ x2 \end{bmatrix}$

Išpjovų sunaudojimas neturi viršyti esamų atsargų. Todėl

```
(%i19) A.X<=at;
```

(%o19) $\begin{bmatrix} 2\,x2+3\,x1 \\ 2\,x1 \\ x2+x1 \\ 2\,x2+x1 \end{bmatrix} <= \begin{bmatrix} 20 \\ 14 \\ 16 \\ 18 \end{bmatrix}$

o apribojimai yra

```
(%i20) apr:[3*x1+2*x2<=20, 2*x1<=14, x1+x2<=16,x1+2*x2<=18, x1>=0,x2>=0];
```

(%o20) $[2\,x2+3\,x1<=20, 2\,x1<=14, x2+x1<=16, 2\,x2+x1<=18, x1>=0, x2>=0]$

```
(%i21) load(simplex);
```

(%o21)
C:/PROGRA~2/MAXIMA~1.2/share/maxima/5.31.2/share/simplex/simplex.mac

```
(%i22) maximize_lp(f,apr);
```

(%o22) $[200, [x2=\frac{17}{2}, x1=1]]$

Kadangi sprendinys turi būti sveikaskaitinis, tai galima manyti, kad reikia pagaminti vieną taburetę ir 8 kėdes.
Bet tai reikia patikrinti. Tada pajamos bus

```
(%i23) 30*1+20*8;
```

(%o23) 190

```
(%i24) load(simplex)$
```

Sveikaskaitiniam sprendimui sudarome programą:

```
(%i25) max_ilp(L,apr):=block([b,r,s,s1,spr,x0,y0],
       maximize_lp(L, apr),
       [x0,y0]:floor(subst(%%[2],[x1,x2])),
       b:makelist(k,k,-10,10),
       s:create_list([x1=x0+i,x2=y0+j],i,b,j,b),
       testas(x):=not member(false,map(is,subst(x,apr))),
       s1:sublist(s,testas),
       spr:[ev(L,s1[1]),s1[1]],
       for k thru length(s1)  do (ev(L,s1[k]),
       if %%>spr[1] then spr:[ev(L,s1[k]),s1[k]]
       elseif %%=spr[1] then spr:append(spr,[s1[k]])),
       spr)$
```

```
(%i26) max_ilp(f,apr);
```

(%o26) $[200,[x1=2,x2=7],[x1=4,x2=4],[x1=6,x2=1]]$

Gavome, kad uždavinys turi tris sveikaskaitinius atsakymus.

3 Dietos uždavinys

Pavyzdys([3], 369 p.) Išspręskime dietos uždavinį, kurio sąlygos pateiktos lentelėje

Figure 1:

Maistingosios medžiagos	Norma	Produktų rūšys	
		P_1	P_2
B_1 – riebalai	4	1	1
B_2 – baltymai	38	6	13
B_3 – angliavandeniai	6	2	1
B_4 – vitaminai	1	0	1
Kaina		7	10

```
(%i27) norma:transpose([4, 38, 6, 1]);
```

(%o27) $\begin{bmatrix} 4 \\ 38 \\ 6 \\ 1 \end{bmatrix}$

Maistingųjų medžiagų kiekiai produktuose išreiškiami matricoje

```
(%i28) A:matrix([1,1],[6,13],[2,1],[0,1]);
```

(%o28) $\begin{bmatrix} 1 & 1 \\ 6 & 13 \\ 2 & 1 \\ 0 & 1 \end{bmatrix}$

Čia A[i,j] - medžigos B[i] kiekis produkte P[j].
Reikia rasti pigiausią maisto mišinio variantą.

Tarkime, kad x1 - produkto P[1] perkamų sąlyginių vienetų skaičius,
x2 - produkto P[2] perkamų sąlyginių vienetų skaičius.
Tada išlaidos išreiškiamos tikslo funkcija

```
(%i29) f:7*x1 + 10*x2;
```

(%o29) $10\, x2+7\, x1$

```
(%i30) X:transpose([x1,x2]);
```

(%o30) $\begin{bmatrix} x1 \\ x2 \end{bmatrix}$

Maistingųjų medžiagų kiekis neturi būti mažesnis negu norma. Todėl

```
(%i31) A.X>=norma;
```

(%o31) $\begin{bmatrix} x2+x1 \\ 13\, x2+6\, x1 \\ x2+2\, x1 \\ x2 \end{bmatrix} >= \begin{bmatrix} 4 \\ 38 \\ 6 \\ 1 \end{bmatrix}$

o apribojimai yra

```
(%i32) apr:[x1+x2>=4, 6*x1+13*x2>=38, 2*x1+x2>=6,x1>=1, x1>=0,x2>=0];
```

(%o32) $[x2+x1>=4, 13\, x2+6\, x1>=38, x2+2\, x1>=6, x1>=1, x1>=0, x2>=0]$

```
(%i33) load(simplex)$
```

```
(%i34) minimize_lp(f,apr);
```

(%o34) $[34, [x2=2, x1=2]]$

4 Mišinių sudarymo uždavinys

Pavyzdys.([2], 96 p.)

Turimi 5 rūšių lyduniai iš švino, cinko ir alavo. Šių medžiagų proporcijos lydiniuose, taip pat 1 kg lydinio kaina yra pateikta lentelėje:

Figure 2:

	Metalo % lydinyje				
Metalas	I	II	III	IV	V
Švinas	10	10	40	60	30
Cinkas	10	30	50	30	20
Alavas	80	60	10	10	50
Kaina	4	4,5	5,8	6	7,5

Reikia rasti kokia proporcija lydyti turimus lydinius, kad gautasis mišinys, kurio 1 kg yr 40% švino, 35% cinko ir 25% alavo mažiausiai kainuotų.
Tegu xj - lydinių kiekiai, viename kg, j=1 ... 5. Tada

```
(%i35) load(simplex)$
```

```
(%i36) L1:x1+x2+x3+x4+x5=1;
```

(%o36) $x5+x4+x3+x2+x1=1$

```
(%i37) kaina:[4,4.5,5.8,6,7.5];
```

(%o37) $[4,4.5,5.8,6,7.5]$

```
(%i38) X:transpose([x1,x2,x3,x4,x5])$
```

Reikia minimizuoti funkciją:

```
(%i39) f:kaina.X;
```

(%o39) $7.5\, x5+6\, x4+5.8\, x3+4.5\, x2+4\, x1$

```
(%i40) b:transpose([0.4,0.35,0.25]);
```

(%o40) $\begin{bmatrix} 0.4 \\ 0.35 \\ 0.25 \end{bmatrix}$

```
(%i41) A:matrix([10,10,40,60,30],[10,30,50,30,20],[80,60,10,10,50]);
```

(%o41) $\begin{bmatrix} 10 & 10 & 40 & 60 & 30 \\ 10 & 30 & 50 & 30 & 20 \\ 80 & 60 & 10 & 10 & 50 \end{bmatrix}$

```
(%i42) eq:A.X=100*b;
```

(%o42) $\begin{bmatrix} 30\, x5+60\, x4+40\, x3+10\, x2+10\, x1 \\ 20\, x5+30\, x4+50\, x3+30\, x2+10\, x1 \\ 50\, x5+10\, x4+10\, x3+60\, x2+80\, x1 \end{bmatrix} = \begin{bmatrix} 40.0 \\ 35.0 \\ 25.0 \end{bmatrix}$

Todėl visi apribojimai yra:

```
(%i43) apr:[L1,30*x5+60*x4+40*x3+10*x2+10*x1=40,
              20*x5+30*x4+50*x3+30*x2+10*x1=35,
              50*x5+10*x4+10*x3+60*x2+80*x1 =25,
              x1>=0 ,x2>=0, x3>=0, x4>=0, x5>=0  ]$
```

```
(%i44) minimize_lp(f,apr),nonegative_lp=true;
```

(%o44) $[5.478571428571429, [x5=0, x4=\frac{9}{28}, x3=\frac{13}{28}, x2=0, x1=\frac{3}{14}]]$

```
(%i45) sort(%[2]);
```

(%o45) $[x1=\frac{3}{14}, x2=0, x3=\frac{13}{28}, x4=\frac{9}{28}, x5=0]$

5 Paskolų ėmimo uždavinys.

Pavyzdys.([1], 269 p.) Trys broliai - Jonas, Petras ir Povilas statyboms nori pasiskolinti pinigų(tūkt. Lt.) Jonui reikia 50, Petrui 40, o Povilui - 60. Paskolinti visus 150 gali du bankai, B1 ir B2: pirmasis 80, o antrasis -70, tačiau reikalauja skirtingų palūkanų. Palūkanų normų lentelė tokia(procentais):

```
(%i46) P:matrix([9,12],[15,8],[12,30]);
```

(%o46) $\begin{bmatrix} 9 & 12 \\ 15 & 8 \\ 12 & 30 \end{bmatrix}$

Poreikiai:

```
(%i47) po:[50,40,60];
```

(%o47) [50,40,60]

Atsargos:

```
(%i48) at:[80,70];
```

(%o48) [80,70]

Paskolos planas:

```
(%i49) X: genmatrix(lambda([i,j], x[concat(i,j)]), 3, 2);
```

(%o49) $\begin{bmatrix} x_{11} & x_{12} \\ x_{21} & x_{22} \\ x_{31} & x_{32} \end{bmatrix}$

```
(%i50) apr:[X[1,1]+X[1,2]=50,X[2,1]+X[2,2]=40,X[3,1]+X[3,2]=60,
       X[1,1]+X[2,1]+X[3,1]=80, X[1,2]+X[2,2]+X[3,2]=70];
```

(%o50) $[x_{12}+x_{11}=50, x_{22}+x_{21}=40, x_{32}+x_{31}=60, x_{31}+x_{21}+x_{11}=80, x_{32}+x_{22}+x_{12}=70]$

Sudarome tikslo funkciją

```
(%i51) f:sum(sum((P[i,j]/100)*X[i,j],i,1,3),j,1,2);
```

(%o51) $\frac{3\,x_{32}}{10}+\frac{3\,x_{31}}{25}+\frac{2\,x_{22}}{25}+\frac{3\,x_{21}}{20}+\frac{3\,x_{12}}{25}+\frac{9\,x_{11}}{100}$

```
(%i52) load(simplex)$
```

```
(%i53) spr:minimize_lp(f,apr),nonegative_lp=true,float;
```

(%o53) $[15.8,[x_{32}=0,x_{31}=60,x_{22}=40,x_{21}=0,x_{12}=30,x_{11}=20]]$

```
(%i54) planas:subst(%[2],X);
```

(%o54) $\begin{bmatrix} 20 & 30 \\ 0 & 40 \\ 60 & 0 \end{bmatrix}$

Atsakymas: Jonas turi skolintis(tūkst. Lt) 20 iš pimojo banko ir 30 iš antrojo banko.
Petras turi skolintis 40 iš antrojo banko, o Povilas 60 iš pirmojo banko. Jie iš viso
turės sumokėti 15,8 tūkst. Lt palūkanų.

6 Literatūra

[1] A.Apynis, E.Stankus, Matematikos pagrindai, Vilnius, TEV, 2009
[2] V.Bubelis, T.Medaiskis, A.Morkeliūnas, Operacijų tyrimo įvadas, Vilnius, VU, 2008
[3] V.Pekarskas, A.Pekarskienė, Tiesinės algebros ir analizinės geometrijos elementai, Kaunas, Technologija, 2004