Geometrinis TP ir NP uždavinių sprendimas

A.Domarkas

1 pvz

Rasti min ir max, jei f=2*x1+x2,
x2+x1<=12,2*x1-x2<=12,2*x1-x2>=0,x2+2*x1>=4,x2>=0.

```
(%i90) load(draw)$
```

Naudosime netiesinės optimizacijos programą paketą "nopt" (autorius A.Domarkas)

```
(%i2) load(nopt);
(%o2) C:/Users/Aleksas/maxima/nopt.mac
```

```
(%i3) f:2*x1+x2;
(%o3) x2+2 x1
```

```
(%i4) apr:[x1+x2<=12, 2*x1-x2<=12,2*x1-x2>=0,2*x1+x2>=4,x2>=0];
(%o4) [x2+x1<=12,2 x1-x2<=12,2 x1-x2>=0,x2+2 x1>=4,x2>=0]
```

```
(%i5) s:minimize_nopt(f,apr);
(%o5) [4,[x1=1,x2=2],[x1=2,x2=0]]
```

Paketo "nopt" komanda "minimize_nopt", randa sprendinių aibės visus kraštinius taškus.
"simplex" paketo komanda "minimize_lp" randa tik vieną sprendinį.

```
(%i6) A:[1,2]; B:[2,0];
(%o6) [1,2]
(%o7) [2,0]
```

Minimumo taškus sudaro visi atkarpos AB taškai.

```
(%i8) t*A+(1-t)*B;
(%o8) [t+2(1-t),2 t]
```

```
(%i9) spr_min:expand(t*A+(1-t)*B);
(%o9) [2-t,2 t]
```

kai 0<=t<=1.

```
(%i10) maximize_nopt(f,apr);
(%o10) [20,[x1=8,x2=4]]
```

```
(%i11) set_draw_defaults(
                x_voxel = 30,
                y_voxel = 30,
                xrange = [0,10],
                yrange = [-1,9],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$
```

```
(%i12) sritis:apply("and", apr);
(%o12) x2+x1<=12∧2 x1-x2<=12∧2 x1-x2>=0∧x2+2 x1>=4∧x2>=0
```

Brėžinys su animacija. Sukinė

```
(%i13) with_slider_draw(
               z, makelist(i, i, 4, 20),
               region(sritis, x1, 0, 10, x2, -1, 9),
               key   = string(ev(f,nouns)=z),
               implicit(f=z,x1,0,10,x2, -1, 10));
```

(%t13)



(%o13)

ats. f_max = 20, kai x1=4, x2=8
f_min = 4, kai x1=2-t, x2=2*t, 0<= t <=1.

Sprendimas su simplex:

```
(%i14) load(simplex);
(%o14)
C:/PROGRA~2/MAXIMA~1.2/share/maxima/5.31.2/share/simplex/simplex.mac
```

```
(%i15) minimize_lp(f,apr);
(%o15) [4,[x2=2,x1=1]]
```

```
(%i16) maximize_lp(f,apr);
(%o16) [20,[x2=4,x1=8]]
```

Visus kampinius leistinosios srities taškus galima rasti su paketu nopt:

```
(%i17) minimize_nopt(1,apr);
(%o17) [1,[x1=1,x2=2],[x1=2,x2=0],[x1=4,x2=8],[x1=6,x2=0],[x1
=8,x2=4]]
```

Suradome visus 5 taškus(žr. brėžinį aukščiau)

2 pvz

1 pvz. Rasti min ir max, jei f=x2-x1^2+6*x1,
3*x2+2*x1<=24, x1+2*x2<=15, 3*x1+2*x2<=24,x2<=4,x1>=0,x2>=0.

```
(%i18) load(nopt)$
```

```
(%i19) load(draw)$
```

```
(%i20) f:x2-x1^2+6*x1;
(%o20) x2-x1^2+6 x1
```

```
(%i21) apr:[2*x1+3*x2<=24, x1+2*x2<=15, 3*x1+2*x2<=24,x2<=4,x1>=0,x2>=0];
(%o21) [3 x2+2 x1<=24,2 x2+x1<=15,2 x2+3 x1<=24,x2<=4,x1>=0,x2>=0]
```

```
(%i22) sritis:apply("and", apr);
(%o22) 3 x2+2 x1<=24∧2 x2+x1<=15∧2 x2+3 x1<=24∧x2<=4∧x1>=0∧x2>=0
```

```
(%i23) m:minimize_nopt(f,apr);
(%o23) [-16,[x1=8,x2=0]]
```

```
(%i24) m[1];
(%o24) -16
```

```
(%i25) M:maximize_nopt(f,apr);
(%o25) [13,[x1=3,x2=4]]
```

```
(%i26) M[1];
(%o26) 13
```

```
(%i27) set_draw_defaults(
                x_voxel = 30,
                y_voxel = 30,
                xrange = [0,10],
                yrange = [0,10],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$
```

```
(%i28) with_slider_draw(
              z, makelist(i,i, -16, 13),
       region(sritis, x1, 0, 10, x2, 0, 10),
              key   = string(f=z),
              implicit(f=z,x1,0,10, x2, 0, 10));
```

(%t28)



(%o28)

3 pvz

Rasti funkcijos f=(x1-3)^2+(x2-4)^2 min ir max, kai 3*x1+2*x2>=7, 10*x1-x2<=8, -18*x1+4*x2<=12,x1>=0, x2>=0.

```
(%i29) load(nopt)$
```

```
(%i30) load(draw)$
```

```
(%i31) f:(x1-3)^2+(x2-4)^2;
```

(%o31) $(x2-4)^2+(x1-3)^2$

```
(%i32) apr:[3*x1+2*x2>=7, 10*x1-x2<=8, -18*x1+4*x2<=12,x1>=0,x2>=0];
```

(%o32) $[2\,x2+3\,x1>=7,10\,x1-x2<=8,4\,x2-18\,x1<=12,x1>=0,x2>=0]$

```
(%i33) sritis:apply("and", apr);
```

(%o33) $2\,x2+3\,x1>=7\wedge 10\,x1-x2<=8\wedge 4\,x2-18\,x1<=12\wedge x1>=0\wedge x2>=0$

```
(%i34) m:minimize_nopt(f,apr);
```

(%o34) $[\frac{324}{101},[x1=\frac{123}{101},x2=\frac{422}{101}]]$

```
(%i35) m[1];
```

(%o35) $\frac{324}{101}$

```
(%i36) M:maximize_nopt(f,apr);
(%o36) [65,[x1=2,x2=12]]

(%i37) M[1];
(%o37) 65

(%i38) set_draw_defaults(
                x_voxel = 40,
                y_voxel = 40,
                xrange = [-5,10],
                yrange = [0,15],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$

(%i39) with_slider_draw(
               z, makelist(i,i, [m[1],20,30,40,50,M[1]]),
       region(sritis, x1, -5, 10, x2, 0, 15),
               key   = string(f=z),
               implicit(f=z,x1,-5,10, x2, 0, 15));
```

(%t39)



(%o39)

4 pvz

Rasti min ir max

```
(%i40) load(nopt)$

(%i41) load(draw)$

(%i42) f:(x1-4)^2+(x2-3)^2;
(%o42) (x2-3)^2+(x1-4)^2
```

```
(%i43) apr:[2*x1+3*x2>=6, 3*x1-2*x2<=18, -x1+2*x2<=8,x1>=0,x2>=0];
```

$$(\%o43)\ [3\,x2+2\,x1\geq 6,3\,x1-2\,x2\leq 18,2\,x2-x1\leq 8,x1\geq 0,x2\geq 0]$$

```
(%i44) sritis:apply("and", apr);
```

$$(\%o44)\ 3\,x2+2\,x1\geq 6\wedge 3\,x1-2\,x2\leq 18\wedge 2\,x2-x1\leq 8\wedge x1\geq 0\wedge x2\geq 0$$

```
(%i45) m:minimize_nopt(f,apr);
```

$$(\%o45)\ [0,[x1=4,x2=3]]$$

```
(%i46) m[1];
```

$$(\%o46)\ 0$$

```
(%i47) M:maximize_nopt(f,apr);
```

$$(\%o47)\ [\frac{549}{4},[x1=13,x2=\frac{21}{2}]]$$

```
(%i48) float(%), numer;
```

$$(\%o48)\ [137.25,[x1=13.0,x2=10.5]]$$

```
(%i49) M[1];
```

$$(\%o49)\ \frac{549}{4}$$

```
(%i50) set_draw_defaults(
                x_voxel = 40,
                y_voxel = 40,
                xrange = [-5,15],
                yrange = [0,15],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$
```

```
(%i51) with_slider_draw(
                z, makelist(i,i, [0.1,1,10,20,30,40,50,137.25]),
       region(sritis, x1, -5, 15, x2, 0, 15),
                key   = string(f=z),
                implicit(f=z,x1,-5,15, x2, 0, 15));
rat: replaced -137.25 by -549/4 = -137.25
rat: replaced -0.1 by -1/10 = -0.1
```

(%t51)



(%o51)

5 pvz

f=3*x1+4*x2, x1^2+x2^2<=25, x1*x2>=4, x1>=0, x2>=0.

```
(%i52) load(nopt)$

(%i53) load(draw)$

(%i54) f:3*x1+4*x2;
(%o54) 4 x2+3 x1

(%i55) apr:[x1^2+x2^2<=25,x1*x2>=4,x1>=0,x2>=0];
(%o55) [x2^2+x1^2<=25, x1 x2>=4, x1>=0, x2>=0]

(%i56) sritis:apply("and", apr);
(%o56) x2^2+x1^2<=25 ∧ x1 x2>=4 ∧ x1>=0 ∧ x2>=0

(%i57) set_draw_defaults(
                x_voxel = 40,
                y_voxel = 40,
                xrange = [0,10],
                yrange = [0,10],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$
```

```
(%i58) with_slider_draw(
               z, makelist(i,i, 13,25),
       region(sritis, x1, 0, 10, x2, 0, 10),
               key   = string(f=z),
               implicit(f=z,x1,0,10, x2, 0, 10));
```

(%t58)



(%o58)

Uždavinį skaidome į du uždavinius:

1 uzd.

```
(%i59) apr:[x1^2+x2^2<=25,x1>=0,x2>=0];
(%o59) [x2^2+x1^2<=25,x1>=0,x2>=0]
```

```
(%i60) M:maximize_nopt(f,apr);
(%o60) [25,[x1=3,x2=4]]
```

2 užd.

```
(%i61) apr:[x1*x2>=4,x1>=0,x2>=0];
(%o61) [x1 x2>=4,x1>=0,x2>=0]
```

```
(%i62) m:minimize_nopt(f,apr);
```

(%o62) $[8\sqrt{3},[x1=\frac{4}{\sqrt{3}},x2=\sqrt{3}]]$

```
(%i63) float(%), numer;
(%o63) [13.85640646055102,[x1=2.309401076758503,x2=1.732050807568877
]]
```

Randame visus reikalingus taškus.

```
(%i64) spr:solve([x2^2+x1^2=25,3*x1+4*x2=z],[x1,x2]),rootscontract,ratsimp;
```

$$(\%o64)\ \left[\left[x1=-\frac{4\sqrt{625-z^2}-3z}{25},x2=\frac{3\sqrt{625-z^2}+4z}{25}\right],\left[x1=\frac{4\sqrt{625-z^2}+3z}{25},x2=-\frac{3\sqrt{625-z^2}-4z}{25}\right]\right]$$

Kad gautume liestines, šaknys turi sutapti. Todėl z galima rasti pošaknius prilyginę nuliui.

```
(%i65) solve(625-z^2);
```

$$(\%o65)\ [z=-25,z=25]$$

```
(%i66) subst(%[2],spr);
```

$$(\%o66)\ [[x1=3,x2=4],[x1=3,x2=4]]$$

```
(%i67) spr:solve([x1*x2=4,3*x1+4*x2=z],[x1,x2]),rootscontract,ratsimp;
```

$$(\%o67)\ \left[\left[x1=-\frac{\sqrt{z^2-192}-z}{6},x2=\frac{\sqrt{z^2-192}+z}{8}\right],\left[x1=\frac{\sqrt{z^2-192}+z}{6},x2=-\frac{\sqrt{z^2-192}-z}{8}\right]\right]$$

```
(%i68) solve(z^2-192);
```

$$(\%o68)\ [z=-8\sqrt{3},z=8\sqrt{3}]$$

```
(%i69) subst(%[2],spr);
```

$$(\%o69)\ \left[\left[x1=\frac{4}{\sqrt{3}},x2=\sqrt{3}\right],\left[x1=\frac{4}{\sqrt{3}},x2=\sqrt{3}\right]\right]$$

A.Domarkas

6 pvz

```
(%i70) load(nopt)$
```

```
(%i71) load(draw)$
```

Rasti min

```
(%i72) f:(x-a)^2+(y-b)^2;
```

$$(\%o72)\ (y-b)^2+(x-a)^2$$

```
(%i73) apr:[(x-8)^2+(y-9)^2<=49,x>=2,x<=13,x+y<=24];
```

$$(\%o73)\ [(y-9)^2+(x-8)^2<=49,x>=2,x<=13,y+x<=24]$$

```
(%i74) sritis:apply("and", apr);
```

$$(\%o74)\ (y-9)^2+(x-8)^2<=49\wedge x>=2\wedge x<=13\wedge y+x<=24$$

```
(%i75) [a,b]:[16,14];
(%o75) [16,14]
```

```
(%i76) f:(x-a)^2+(y-b)^2;
(%o76) (y-14)^2+(x-16)^2
```

```
(%i77) s:minimize_nopt(f,apr);
(%o77) [18,[x=13,y=11]]
```

```
(%i78) set_draw_defaults(
                x_voxel = 30,
                y_voxel = 30,
                xrange = [0,35],
                yrange = [0,25],
                grid   = true,
                proportional_axes = xy,
                fill_color = skyblue)$
```

```
(%i79) with_slider_draw(
               z, makelist(s[1]+2*i, i, 0, 10),
       region(sritis, x, 0, 35, y, 0, 25),
               key   = string(f=z),
               implicit(f=z,x,0,35,y, 0, 25));
```

(%t79)



(%o79)

Sprendinys [x=13, y=11] yra srities kampe

7 pvz

```
(%i80) [a,b]:[11,10];
(%o80) [11,10]
```

Rasti min

```
(%i81) f:(x-a)^2+(y-b)^2;
(%o81) (y-10)^2+(x-11)^2

(%i82) apr:[(x-8)^2+(y-9)^2<=49,x>=2,x<=13,x+y<=24];
(%o82) [(y-9)^2+(x-8)^2<=49,x>=2,x<=13,y+x<=24]

(%i83) s:minimize_nopt(f,apr);
(%o83) [0,[x=11,y=10]]

(%i84) with_slider_draw(
               z, makelist(s[1]+i, i, 1, 8),
               region(sritis, x, 0, 35, y, 0, 25),
               key   = string(ev(f,nouns)=z),
               implicit(f=z,x,0,35,y, 0, 25)),a=11,b=10;
```

(%t84)



(%o84)

Sprendinys [x=11, y=10] yra srities viduje

8 pvz

```
(%i85) [a,b]:[14,14];
(%o85) [14,14]

(%i86) f:(x-a)^2+(y-b)^2;
(%o86) (y-14)^2+(x-14)^2

(%i87) apr:[(x-8)^2+(y-9)^2<=49,x>=2,x<=13,x+y<=24];
(%o87) [(y-9)^2+(x-8)^2<=49,x>=2,x<=13,y+x<=24]

(%i88) s:minimize_nopt(f,apr),a=14,b=14;
(%o88) [8,[x=12,y=12]]
```

```
(%i89) with_slider_draw(
               z, makelist(s[1]+i, i, 0, 8),
               region(sritis, x, 0, 35, y, 0, 25),
               key   = string(ev(f,nouns)=z),
               implicit(f=z,x,0,35,y, 0, 25)),a=14,b=14;
```

(%t89)



(%o89)

Sprendinys [x=12, y=12] yra ant kraštinės

Literatūra:
1. A.Apynis, V.Stankus, Matematika, Vilnius, TEV, 2001