Dviejų kintamųjų funkcijų ekstremumai

A.Domarkas, VU

Čia naudojame atvirojo kodo CAS Maxima 5.41.0
Maxima build date: 2017-10-03

Apibrėžimas.
Taškas (a,b) yra vadinamas funkcijos z = f(x,y) lokalaus maksimumo tašku,
jei f(x,y) ≤ f(a,b) su visais (x,y) iš taško (a,b) aplinkos.
Taškas (a,b) yra vadinamas funkcijos f(x,y) lokalaus minimumo tašku,
jei f(x,y) ≥ f(a,b) su visais (x,y) iš taško (a,b) aplinkos.
Lokalaus minimumo ir lokalaus maksimumo taškai yra vadinami lokalaus
ekstremumo taškais.

Apibrėžimas.
Taškas (a,b), priklausantis funkcijos z = f(x,y) apibrėžimo sričiai,
vadinamas kritiniu arba stacionariuoju tašku, jei abi dalinės išvestinės
tame taške lygios nuliui arba bent viena neegzistuoja.

1 teorema. Būtinoji lokalaus ekstremumo sąlyga.
Jei (a,b) yra funkcijos z = f(x,y) lokalaus ekstremumo taškas ir tame
taške egzistuoja dalinės išvestinės, tai jos tame taške lygios nuliui:
f'_x(a,b) = 0 ir f'_y(a,b) = 0.

2 teorema. Pakankamoji lokalaus ekstremumo sąlyga.
Tegul taškas (a,b) yra funkcijos z = f(x,y) kritinis taškas ir šio taško
aplinkoje egzistuoja tolydžiosios antrosios išvestinės. Apibrėžkime
A = f''_xx(a,b) ir D = f''_xx(a,b)f''_yy(a,b)- f''_xy(a,b)^2.

Tada:
1. jei D > 0 ir A < 0, tai (a,b) yra lokalaus maksimumo taškas;
2. jei D > 0 ir A > 0, tai (a,b) yra lokalaus minimumo taškas;
3. jei D < 0, tai taške (a,b) ekstremumo nėra, (a,b) yra balno taškas;
4. jei D = 0, tai šis požymis nepritaikomas - reikia papildomo tyrimo.

Kritiniams taškams tirti apibrėžiame komandą "testas":

```
(%i1)  testas(kt):=block(
       [A,B,C]:subst(kt,[diff(f,x,2),diff(f,x,1,y,1),diff(f,y,2)]),
       D: A*C-B^2,
       if D>0 and A<0 then print("Taške ",kt,maksimumas=subst(kt,f))
       elseif D>0 and A>0 then print("Taške ",kt,minimumas=subst(kt,f))
       elseif D<0 then print("Taške ",kt,"ekstremumo nėra(balno taškas)")
       elseif D=0 then print("Taške ",kt,"reikia papildomo tyrimo"))$
```

Žr. [1], 210 p.; [2], 218 p.; [3]

Pavyzdys(Žr. [2]). Rasime funkcijos f ekstremumus.

```
(%i2)  f:x^3+y^3-3*x^2-3*y^2-9*x;
```

(f) $y^3-3\,y^2+x^3-3\,x^2-9\,x$

```
(%i4)  L1:diff(f,x)=0; L2:diff(f,y)=0;
```

(L1) $3\,x^2-6\,x-9=0$

(L2) $3\,y^2-6\,y=0$

Randame kritinius taškus:

```
(%i5)  krit_taskai:solve([L1,L2],[x,y]);
```

(krit_taskai) $[[x=3,y=0],[x=-1,y=0],[x=3,y=2],[x=-1,y=2]]$

```
(%i6)  map(testas,krit_taskai)$
```

Taške $[x=3,y=0]$ *ekstremumo nėra(balno taškas)*
Taške $[x=-1,y=0]$ *maksimumas* $=5$
Taške $[x=3,y=2]$ *minimumas* $=-31$
Taške $[x=-1,y=2]$ *ekstremumo nėra(balno taškas)*

(%i7) `wxcontour_plot(f, [x, -2, 4], [y, -1.5, 3.5], [legend,false], [gnuplot_preamble,"set cntrparam levels 24"])$`

(%t7)



2 pavyzdys. (Žr. [3])

(%i8) `f:(x+y)*(x*y+x*y^2);`

(f) $(y+x)\,(x\,y^2+x\,y)$

(%i10) `L1:diff(f,x)=0; L2:diff(f,y)=0;`

(L1) $(y+x)\,(y^2+y)+x\,y^2+x\,y=0$

(L2) $x\,y^2+(y+x)\,(2\,x\,y+x)+x\,y=0$

Randame kritinius taškus:

(%i11) `krit_taskai:solve([L1,L2],[x,y]);`

(krit_taskai) $[[x=0,y=0],[x=0,y=-1],[x=\frac{3}{8},y=-\frac{3}{4}],[x=1,y=-1]]$

(%i12) `map(testas,krit_taskai)$`

Taške $[x=0,y=0]$ *reikia papildomo tyrimo*

Taške $[x=0,y=-1]$ *ekstremumo nėra(balno taškas)*

Taške $[x=\frac{3}{8},y=-\frac{3}{4}]$ *maksimumas*$=\frac{27}{1024}$

Taške $[x=1,y=-1]$ *ekstremumo nėra(balno taškas)*

(%i13) `wxcontour_plot(f, [x, 0, 1], [y, -1, 0], [legend,false], [gnuplot_preamble,"set cntrparam levels 36"])$`

(%t13)



3 pavyzdys. Rasime Himmelblau funkcijos([4]) ekstremumus.

(%i14) `f:(x^2+y-11)^2+(x+y^2-7)^2;`

(f) $(y^2+x-7)^2+(y+x^2-11)^2$

(%i16) `L1:diff(f,x)=0; L2:diff(f,y)=0;`

(L1) $2\,(y^2+x-7)+4\,x\,(y+x^2-11)=0$

(L2) $4\,y\,(y^2+x-7)+2\,(y+x^2-11)=0$

Randame kritinius taškus:

(%i17) `krit_taskai:solve([L1,L2],[x,y]);`

(krit_taskai) $[[x=3.584428223844282, y=-1.848126535626536], [x=-2.80511811023622, y=3.131312515247621], [x=-3.779310344827586, y=-3.283185840707965], [x=3, y=2], [x=0.08667750491999658, y=2.884254431699687], [x=-0.1279613466334165, y=-1.953714981729598], [x=-0.2708445885107149, y=-0.9230384807596201], [x=-3.0730257571469, y=-0.08135304590138556], [x=3.38515406162465, y=0.07385188249896565]]$

(%i18) `length(%);`

(%o18) 9

Gavome, kad fukcija turi 9 kritinius taškus. Jų klasifikacijai pritaikome komandą "testas".

(%i19) `map(testas,krit_taskai)$`

Taške $[x=3.584428223844282, y=-1.848126535626536]$ *minimumas* $=7.190238265563076\ 10^{-13}$
Taške $[x=-2.80511811023622, y=3.131312515247621]$ *minimumas* $=1.805914487428136\ 10^{-14}$
Taške $[x=-3.779310344827586, y=-3.283185840707965]$ *minimumas* $=1.875485921124983\ 10^{-12}$
Taške $[x=3, y=2]$ *minimumas* $=0$
Taške $[x=0.08667750491999658, y=2.884254431699687]$ *ekstremumo nėra(balno taškas)*
Taške $[x=-0.1279613466334165, y=-1.953714981729598]$ *ekstremumo nėra(balno taškas)*
Taške $[x=-0.2708445885107149, y=-0.9230384807596201]$ *maksimumas* $=181.6165215225827$
Taške $[x=-3.0730257571469, y=-0.08135304590138556]$ *ekstremumo nėra(balno taškas)*
Taške $[x=3.38515406162465, y=0.07385188249896565]$ *ekstremumo nėra(balno taškas)*

Brėžime lygio linijas vienintelio maksimumo taško [x=-0.27084458851071,y=-0.92303848075962] aplinkoje:

(%i20) `wxcontour_plot(f, [x,-0.3,-0.24], [y,-1,-0.85], [legend,false], [gnuplot_preamble,"set cntrparam levels 12"])$`

(%t20)



Geriau negu su "contour_plot" nubrėžti lygio linijas galima su paketu "implicit_plot". Tada lygių reikšmes reikia parinkti patiems.

(%i21) `load(implicit_plot)$`

(%i22) wximplicit_plot([f=1,f=10,f=20,f=30,f=160,f=180],
[x,-5,5],[y,-5,5],[legend,false]);

(%t22) 

(%o22)

Dar kartą randame visus penkis lokaliuosius ekstremumus su paketu "fmin_cobyla":

(%i23) load(fmin_cobyla)$

```
 0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
```

(%i24) fmin_cobyla(f, [x, y], [-3,3], constraints = [], iprint=0);

(%o24) $[[x=-2.805117714454949, y=3.131311546072143], 4.204711597294184\ 10^{-11}, 55, 0]$

(%i25) fmin_cobyla(f, [x, y], [-3,-3], constraints = [], iprint=0);

(%o25) $[[x=-3.779310588699752, y=-3.283186262162302], 7.207582117009756\ 10^{-12}, 51, 0]$

(%i26) fmin_cobyla(f, [x, y], [1,1], constraints = [], iprint=0);

(%o26) $[[x=3.000001489038577, y=1.999998519499293], 7.5209287826238\ 10^{-11}, 51, 0]$

(%i27) fmin_cobyla(f, [x, y], [1,-2], constraints = [], iprint=0);

(%o27) $[[x=3.584427315412266, y=-1.84812815631613], 1.055595151567217\ 10^{-10}, 56, 0]$

(%i28) fmin_cobyla(-f, [x, y], [-1,-1], constraints = [], iprint=0);

(%o28) $[[x=-0.2708453030493918, y=-0.923038169769575], -181.6165215225714, 53, 0]$

4 pavyzdys

(%i29) f:(x-1/2)^2*(x+1)^2+(y+1)^2*(y-1)^2;

(f) $(y-1)^2 (y+1)^2+\left(x-\frac{1}{2}\right)^2 (x+1)^2$

(%i31) L1:diff(f,x)=0; L2:diff(f,y)=0;

(L1) $2\left(x-\frac{1}{2}\right)(x+1)^2+2\left(x-\frac{1}{2}\right)^2 (x+1)=0$

(L2) $2(y-1)(y+1)^2+2(y-1)^2(y+1)=0$

Randame kritinius taškus:

(%i32) krit_taskai:solve([L1,L2],[x,y]);

(krit_taskai) $[[x=\frac{1}{2}, y=0], [x=-1, y=0], [x=-\frac{1}{4}, y=0], [x=\frac{1}{2}, y=1], [x=-1, y=1], [x=-\frac{1}{4}, y=1], [x=\frac{1}{2}, y=-1], [x=-1, y=-1], [x=-\frac{1}{4}, y=-1$
$]]$

(%i33) length(%);

(%o33) 9

Gavome, kad fukcija turi 9 kritinius taškus. Jų klasifikacijai pritaikome komandą "testas".

(%i34) `map(testas,krit_taskai)$`

Taške $[x=\frac{1}{2}, y=0]$ *ekstremumo nėra(balno taškas)*

Taške $[x=-1, y=0]$ *ekstremumo nėra(balno taškas)*

Taške $[x=-\frac{1}{4}, y=0]$ *maksimumas* $=\frac{337}{256}$

Taške $[x=\frac{1}{2}, y=1]$ *minimumas* $=0$

Taške $[x=-1, y=1]$ *minimumas* $=0$

Taške $[x=-\frac{1}{4}, y=1]$ *ekstremumo nėra(balno taškas)*

Taške $[x=\frac{1}{2}, y=-1]$ *minimumas* $=0$

Taške $[x=-1, y=-1]$ *minimumas* $=0$

Taške $[x=-\frac{1}{4}, y=-1]$ *ekstremumo nėra(balno taškas)*

(%i35) `load(implicit_plot)$`

(%i36) `wximplicit_plot([f=1/8,f=1/4,f=1/2,f=1,f=337/256-0.1,f=2,f=3],`
`[x,-2,1.5],[y,-2,2],[legend,false]);`

```
 rat: replac
ed 1.21640625 by 1557/1280 = 1.21640625
```

(%t36)



(%o36)

(%i37) `337/256;`

(%o37) $\frac{337}{256}$

(%i38) `float(%), numer;`

(%o38) 1.31640625

[1] http://tutorial.math.lamar.edu/Classes/CalcIII/RelativeExtrema.aspx
[2] https://en.wikipedia.org/wiki/Second_partial_derivative_test
[3] http://mathworld.wolfram.com/SecondDerivativeTest.html
[4] http://en.wikipedia.org/wiki/Himmelblau%27s_function